

Autolev Tutorial

OnLine Dynamics, Inc.
1605 Honfleur Drive
Sunnyvale, CA 94087 USA
Phone: 408-736-9566
Web: http://www.autolev.com
E-mail: info@autolev.com

Last updated September 29, 2003 for `Autolev` version 4

Contents

1 Getting Started

1.1 Running Autolev

- On a Windows or Macintosh computer, double click on the `Autolev` icon
- On a Windows computer, `cd` to the directory containing the file "al.exe" and type `al`
- On a Unix or Linux computer, `cd` to the directory containing the file "al" and type `./al`

When `Autolev` starts, the following prompt is displayed:

```
(1)
```

`Autolev` is case-insensitive. Thus "apple", "Apple", and "APPLE" are equivalent. Depending on the command entered, `Autolev` may or may not output a response. `Autolev` distinguishes between input and output lines by preceding output lines with an arrow (->). Try the following example.

```
   (1) Constants A, B
   (2) X = 4*(A+B) - 3*(a-b)
-> (3) X = A + 7*B
```

1.2 Online Help

Typing `WHAT` at an `Autolev` prompt causes an alphabetical list of commands to be displayed on the screen. To obtain help in connection with a particular command on the list, type `HELP commandname`. For example, type `HELP SAVE` for more information on `SAVE`.

1.3 The Save Command

Issuing a `SAVE` command causes a text file dealing with the current session to be created. Two of the six syntaxes of the `SAVE` command are

```
SAVE filename.al
SAVE filename.all
```

The first causes all input lines to be stored in `filename.al`, whereas the second causes both input and response lines to be stored in `filename.all`. The file `filename.al` may be subsequently edited with a text editor and re-run as described in the next section.

1.4 Autolev Input Files

Instead of entering `Autolev` commands interactively, one can use a text editor to create a text file, e.g., `michele.al`, and then execute the commands in `michele.al` by:

- invoking `Autolev` and typing `RUN michele.al` at an `Autolev` line prompt
- typing `al michele.al` at the operating system prompt (Windows, Unix, Linux)
- dragging and dropping the file `michele.al` on top of the `Autolev` icon (Windows)

One advantage of execution by reading from a text file is that commands may be broken into multiple lines. An `Autolev` input line cannot contain more than 256 characters, but lines of unlimited length may be entered, provided that an ampersand `(&)` appear as the last non-blank character of each but the last input line, informing `Autolev` that the command continues on the next line.

1.5 Running the `Alplot` plotting program

To invoke `Alplot` on a Windows machine, double click on the `Alplot` icon and follow the on-screen instructions. The Windows version of `Alplot` supports drag and drop, i.e., drag a data file on top of the `Alplot` icon to start `Alplot` and load the data file. Alternately, the Windows version of `Alplot` can be invoked by typing one of the following syntaxes at the operating system prompt:

`alplot`	Invokes `alplot`
`alplot filename`	Invokes `alplot` and loads the data file `filename`
`alplot filenameA filenameB ...`	Invokes `alplot` and loads the data files `filenameA`, `filenameB`, ...
`alplot filenameA filenameA`	Invokes `alplot` and loads the data file `filenameA`, twice

To invoke `Alplot` on a Unix or Linux machine, type one of the following syntaxes at the operating system prompt:

`alplot`	Invokes `alplot`
`alplot filename`	Invokes `alplot` with the datafile `filename`
`alplot filename T,X,Y`	Invokes `alplot` with the datafile `filename` and plots the columns with headings `X` and `Y` versus the column with heading `T`
`alplot filename 1,2,4`	Invokes `alplot` with the datafile `filename` and plots column 2 and column 4 versus column 1

2 Mathematical Entities

2.1 Scalars

The names of scalar quantities must start with a letter. This letter may be followed by alphanumeric characters or underscores (_). For example, `XYZ`, `AB3`, and `ABC_3` are acceptable names. 1^{st}, 2^{nd}, 3^{rd}, ..., ordinary derivatives of a scalar variable with respect to `T` are denoted with primes ('). The prime appear at the end of the name, and each prime represents one differentiation, e.g., `X''` is the second ordinary derivative of `X` with respect to `T`. At most 64 characters, including primes, may be used to form the name of a scalar quantity. Before a scalar quantity may be used in an analysis, it must be either named in a `Constants`, `Variables`, `Specified`, `Imaginary`, `Mass`, or `Inertia` declaration, or defined by appearing on the left-hand side of an equals sign in an assignment. For example,

```
Constants  A,B=3  % Declares the constants A and B, and assigns the value 3 to B
Constants  C+     % Declares C to be a non-negative constant
Constants  D-     % Declares D to be a non-positive constant
Specified  Phi    % Declares PHI as a function of time, constants, and variables
Variables  V, W   % Declares the variables V and W
Variables  X''    % Declares the variables X, X', and X''
Variables  Y{3}'  % Declares the variables Y1, Y2, Y3, Y1', Y2', and Y3'
Variables  U{3}   % Declares the generalized speeds U1, U2, and U3
Variables  U{1}'  % Declares the generalized speed U1 and its time derivative U1'
Imaginary  j      % Declares j to be the imaginary number, i.e., sqrt(-1)
Tina = 2*t        % Creates the quantity Tina and assigns the value 2*t to Tina
```

2.2 Vectors, Dyadics, and Polyadics

To distinguish scalars, vectors, dyadics, and polyadics from each other, one "greater than" > is appended to the end of a vector's name, two are appended to the end of a dyadic's name, and three to the end of the name of a triadic or higher-order polyadic. The names of vectors, dyadics, and other polyadics must start with a letter. This letter may be followed by alphanumeric characters or

underscores (_). At most 64 characters, including > symbols, may be used to form the name of a vector, dyadic, or higher order polyadic. Examples of names are

`ed>`, `ed1>`	vectors
`john>>`, `john_paul>>`	dyadics
`tom>>>`	triadic or higher order polyadic

2.3 Matrices

Matrices start with a left bracket (`[`) and end with a right bracket (`]`). Elements of a row are separated from each other by a comma, while rows are separated by semicolons. For example, `[1,2,3;4,5,6]` denotes the matrix $\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix}$. The name of a matrix must start with a letter. This letter may be followed by alphanumeric characters or underscores (_). At most 64 characters may be used to form the name of a matrix.

The name of an element of a matrix consists of the name of the matrix, followed by a left bracket (`[`), an expression that must evaluate to a positive integer, a comma, another expression which must evaluate to a positive integer, and a right bracket (`]`). If a matrix is one-dimensional, then one may refer to its elements in a shorthand form, namely, the name of a matrix followed by a left bracket (`[`), an expression that evaluates to a positive integer, and a right bracket (`]`). For example, `X[7]` denotes the 7^{th} element of the matrix `X`, regardless of whether `X` is a row matrix or a column matrix.

2.4 Reserved Names

- `T` is a reserved variable, often used to denote time.
- `PI` is a reserved constant with the value of 3.1415...
- `0>` is the zero vector.
- `1>>` is the unit dyadic.
- Unless otherwise declared, `i =` $\sqrt{-1}$.
- The declaration `Variables U{n}` or `Variables U{n}'` introduces generalized speeds.
- `U_INDEPENDENT`, `U_INDEPENDENT'`, `U_DEPENDENT`, `U_DEPENDENT'`, `U_AUXILIARY`, `U_AUXILIARY'`, `U_PARTIALS`, and `U_PARTIALS'` are reserved.
  Type `HELP Constrain` for details.
- `ZERO` is reserved for the matrix of expressions which one sets equal to zero to form dynamical equations of motion.
- `DEPENDENT` and `AUXILIARY` are reserved for matrices of expressions which one sets equal to zero to form motion constraint equations.
- `ZEE_NOT` is reserved for a matrix of scalar quantities that should be excluded from `Z1,Z2,...`.

3 Physical Entities

3.1 Bodies, Frames, Newtonian, Particles, Points

A Newtonian reference frame must be named in the `Newtonian` declaration. The names of bodies, frames, particles, and points must appear, respectively, in `Bodies`, `Frames`, `Particles`, and `Points` declarations, and these names must begin with a letter followed by alphanumeric characters (no underscores). Each name may consist of at most 27 characters.

```
Bodies    A    % Declares the (massive) rigid body A.
               % Introduces orthonormal vectors A1>, A2>, A3> fixed in A.
               % Declares a point Ao that is coincident with A's mass center
Frames    B    % Declares the (massless) reference frame B.
               % Introduces orthonormal vectors B1>, B2>, B3> fixed in B.
               % Declares a point Bo that is fixed on B
Newtonian N    % Declares N as a Newtonian (inertial) reference frame.
               % Introduces orthonormal vectors N1>, N2>, N3> fixed in N.
               % Declares a point No that is fixed on N
Particles C,D  % Declares the (massive) particles C and D
Points    E,F  % Declares the (massless) points E and F
```

3.2 Syntactical Forms

The underscore (_) separates points and/or reference frames in the names of position vectors, velocities, accelerations, direction cosine matrices, angular velocities, angular accelerations, forces, torques, and inertia dyadics. Examples follow.

```
P_O_Q>     % Position vector from point O to point Q
V_P_N>     % Velocity of point P in reference frame N
A_D_C>     % Acceleration of point D in reference frame C
W_B_F>     % Angular velocity of body B in reference frame F
ALF_E_A>   % Angular acceleration of body E in reference frame A
```

```
Force_P>   % Force acting on point P
Torque_B>  % Torque of a couple acting on body B
I_B_P>>    % Inertia dyadic of body B for point P
A_B        % Direction cosine matrix relating A1>,A2>,A3> to B1>,B2>,B3>
```

Note: Element `A_B[i,j]` of `A_B` is defined as `DOT(Ai>,Bj>) (i,j=1,2,3)`. As a consequence,

$$\texttt{A\_B} = \begin{bmatrix} A1> \cdot B1> & A1> \cdot B2> & A1> \cdot B3> \\ A2> \cdot B1> & A2> \cdot B2> & A2> \cdot B3> \\ A3> \cdot B1> & A3> \cdot B2> & A3> \cdot B3> \end{bmatrix}$$

3.3 Mass Declarations

The masses of bodies and particles are specified by the `Mass` declaration. For example, consider the following `Autolev` file:

```
   (1) Bodies     A,B
   (2) Particles  C
   (3) Mass       A=12.5, B=MB, C=MC=10
-> (4) MC = 10
   (5) Mass_A = Mass(A)          % Returns the mass of A
-> (6) Mass_A = 12.5
   (7) Mass_AB = Mass(A,B)       % Returns Mass(A) + Mass(B)
-> (8) Mass_AB = 12.5 + MB
   (9) TOTAL  = Mass()           % Returns Mass(A) + Mass(B) + Mass(C)
-> (10) TOTAL = 22.5 + MB
```

After `A`, `B`, and `C` have been declared as bodies or particles, the `Mass` declaration specifies that `A` has a mass of `12.5`, `B` has a mass of `MB`, and `C` has a mass of `MC` where `MC=10`. Once the mass of a body or particle has been declared, its mass can be referenced by typing `Mass(NAME)`.

3.4 Inertia Declarations

As previously mentioned, `I_B_P>>` denotes the inertia dyadic of body B for point P. Inertia dyadics may be created in the following three ways:

- By explicit assignment. To assign a dyadic to `I_B_P>>`, type, for example,

  ```
  I_B_P>> = 100*a1>*a1> + 200*a2>*a2> + 250*a3>*a3>
  ```

- With one of the variants of the `Inertia` declaration. For example, typing

  ```
  Inertia  A, I11,I22,I33,I12,I23,I31
  ```

is equivalent to typing

```
Constants  I11+, I22+, I33+, I12, I23, I31
I_A_Ao>> = I11*A1>*A1> + I12*A1>*A2> + I31*A1>*A3>
         + I12*A2>*A1> + I22*A2>*A2> + I23*A2>*A3>
         + I31*A3>*A1> + I23*A3>*A2> + I33*A3>*A3>
```

Any inertia scalar that is not specified is set equal to zero by default.

- If the inertia dyadic of a body for a point has been entered, `Autolev` can find the inertia dyadic of the body for a second point if the mass of the body has been declared and appropriate position vectors have been entered. For example,

```
   (1) Points P
   (2) Bodies B
   (3) Mass B=MB
   (4) Constants I,J,L
   (5) P_Bo_P> = L*B1>
-> (6) P_BO_P> = L*B1>
   (7) I_B_P>> = I*B1>*B1> + I*B2>*B2> + J*B3>*B3>
-> (8) I_B_P>> = I*B1>*B1> + I*B2>*B2> + J*B3>*B3>
   (9) I_B_BO>> = Inertia(Bo,B)
-> (10) I_B_BO>> = I*B1>*B1> + (I-MB*L^2)*B2>*B2> + (J-MB*L^2)*B3>*B3>
```

3.5 Forces and Torques

`Force_P>` is the force acting on a point or particle `P`, and `Torque_B>` is the torque of the couple acting on a frame or body B. There are four ways to assign values to a `Force` or `Torque` vector:

- `Force_Q> = 7*a1>  % Explicit assignment overwrites previous value for Force_Q>`
- `Force_P> += 5*a3>  % Adds 5*a3> to the value of Force_P>`
- `Force_P> -= 5*a3>  % Subtracts 5*a3> from the value of Force_P>`
- `Force(P/Q,VEC>)  % Equivalent to  Force_P> -= VEC>`
  `Force_Q> += VEC>`
- `Torque_B> = 7*B3>  % Explicit assignment overwrites previous value for Torque_B>`
- `Torque_A> += VEC>  % Adds VEC> to the value of Torque_A>`
- `Torque_A> -= VEC>  % Subtracts VEC> from the value of Torque_A>`
- `Torque(A/B,VEC>)  % Equivalent to  Torque_A> -= VEC>`
  `Torque_B> += VEC>`

4 Procedures for Solving Problems

4.1 Numerical Integration of Differential Equations

Autolev writes Matlab, C, or Fortran code for the numerical integration of sets of differential equations. The differential equations can be either dynamical equations of motion generated by Autolev or differential equations entered directly by the user. The procedure for generating Matlab, C, or Fortran codes and their input files is summarized below.

- Declare all variables and their derivatives.
- Enter the equations governing the highest derivative of each variable.
- Use Input statements to specify the values of constants, the initial values of variables, and values of integration parameters such as tInitial, tFinal, and integStp.
- Use Output statements to specify the quantities to be output.
- Type CODE ODE() FILE.EXT (EXT is m, c, for, or f).
  When EXT is c, this creates ready-to-compile C code in FILE.c and an input file FILE.IN.
  When EXT is f or for, this creates Fortran code in FILE.EXT and an input file FILE.IN.
  When EXT is m, this creates ready-to-run Matlab code in FILE.m

```
% Example - numerically integrate the following differential equations:
% dx/dt = -a*x - b*y
% dy/dt = -y - t*x^2
%         where a and b are constants and t is time
Constants a,b
Variables x', y'
x' = -a*x - b*y
y' = -y - t*x^2
Input a=1, b=2, x=2, y=3
Input tInitial=0, tFinal=5, integStp=0.1
Output t,x,y,x',y'
CODE ODE() odexy.c
```

This creates ready-to-compile C code in the file odexy.c and an input file odexy.in that is read by the executable program, e.g., odexy.exe. The executable program performs numerical integration of the differential equations for x and y from t=0 to t=5 with integration steps of 0.1.

4.2 Numerical Solution of Nonlinear Nondifferential Equations

The procedure for writing Matlab, C, or Fortran code for the numerical solution of sets of nonlinear algebraic equations is as follows:

- Declare all variables.
- Create a matrix whose elements represent the nonlinear equations.
- Use Input statements to specify guesses for the values of the variables, the values of constants, and values of convergence parameters such as absErr and relErr.
- Type CODE Nonlinear(matrixName,variables) FILE.EXT (EXT is m, c, for, or f).

```
% Example - numerically solve the following equations for x and y:
%   x^2 + y^2 - r = 0
%   y - a*sin(x) = 0
Variables  x, y
Constants  a, r
eqn[1] = x^2 + y^2 - r              % Equation of a circle
eqn[2] = y - a*sin(x)               % Equation of a sinusoid
Input a=1.0, r=1.0
Input x=0.5, y=0.5
CODE Nonlinear(eqn, x,y) katy.m
```

This creates the ready-to-run Matlab file katy.m. To solve the nonlinear equations, invoke Matlab and type katy at the Matlab prompt. To try a different initial guess for x or y or to use a different value for a or b, use a text editor to edit katy.m.

4.3 Numerical Solution of Linear Algebraic Equations

The procedure for writing Matlab, C, or Fortran code for the numerical solution of sets of linear equations is as follows:

- Declare all variables.
- Create a matrix whose elements represent the linear equations.
- Use Input statements to specify the the values of constants.
- Type CODE Algebraic(matrixName,variables) FILE.EXT (EXT is m, c, for, or f).

```
% Example - numerically solve the following equations for x and y:
%   a11*x + a12*y = b1
%   a21*x + a22*y = b2
Variables  x, y
Constants  a{1:2,1:2}, b{1:2}
eqn[1] = a11*x + a12*y - b1
eqn[2] = a21*x + a22*y - b2
Input a11=1, a12=2, a21=3, a22=4, b1=5, b2=6
CODE Algebraic(eqn, x,y) becky.for
```

This creates ready-to-compile Fortran code in the file `becky.for` and an input file `becky.in` that is read by the executable program, e.g., `becky.exe`. The executable program solves the linear equations for `x` and `y` for the given values of `a11`, `a12`, `a21`, `a22`. To solve for `x` and `y` with different values of `a11`, `a12`, `a21`, `a22`. use a text editor to edit `becky.in` and then re-run `becky.exe`.

4.4 Equations of Motion

Although `Autolev` can assist one in formulating equations of motion by any method, it is especially well suited for carrying out dynamic analyses with Kane's method (see `Dynamics Online: Theory and Implementation with Autolev`™ by T. R. Kane and D. A. Levinson). Examples are presented in Chapter 5 of this tutorial. The procedure for generating equations of motion via Kane's method is summarized below.

- Declare a `Newtonian` reference frame
- Declare `Bodies`, `Frames`, `Points`, and `Particles`
- Declare generalized speeds and their time-derivatives with a declaration of the form
  `Variables U{n}' % n is a positive integer`
- Declare generalized coordinates and their time-derivatives with a `Variables` declaration
- Create kinematical differential equations by relating the time-derivatives of generalized coordinates to the generalized speeds, e.g., `X' = U1*COS(Q) + U2*SIN(X)`
- Form position vectors and direction cosine matrices
- Form angular and linear velocities
- If necessary, impose motion constraints with the `Constrain` command
- Form angular and linear accelerations. `Autolev` can *automatically* form angular and linear accelerations by differentiation of corresponding angular and linear velocities. For large problems, or problems which require real-time solution, form accelerations with the `A2pts` and `A1pt` commands and turn `AutoZ ON`.
- Enter expressions for forces and torques
- Form equations of motion by entering `ZERO = Fr() + FrStar()`
- Bring the equations of motion into a form suitable for motion simulation by issuing the `Kane()` command

4.5 Motion Constraints

One function of the `Constrain` command is to solve constraint equations for dependent generalized speeds when a system is subject to motion constraints. Additionally, if the time-derivatives of generalized speeds are declared, the `Constrain` command produces expressions for the time-derivatives of the dependent generalized speeds. For more information about the `Constrain` command, type `HELP Constrain` at the `Autolev` line prompt, and/or see `Dynamics Online: Theory and Implementation with Autolev`™. Examples are presented in Chapter 5 of this tutorial. A summary of the procedure follows.

- Declare as many generalized speeds as necessary to characterize the motion of the unconstrained system.
- Construct m expressions such that setting the expressions equal to zero produces the motion constraint equations. Assign these expressions to the first m elements of the `DEPENDENT` matrix. For example,

```
DEPENDENT[1] = U2 - U1
DEPENDENT[2] = U3 + U2
```

- Type `Constrain( DEPENDENT[ Ui, Uj,...] )`
  where `Ui, Uj, ...` are the names of the m dependent generalized speeds.
  The `Constrain` command solves the m equations associated with the `DEPENDENT` matrix for the m dependent generalized speeds, and if the time-derivatives of the dependent generalized speeds have been declared, the `Constrain` command solves for those also.

```
   (1) %  Problem:  Triple pendulum whose tip has a specified motion
   (2) Newtonian    N               % Newtonian reference frame
   (3) Bodies       A, B, C         % Links of pendulum
   (4) Points       P               % Distal end of C
   (5) Variables    U{3}            % Generalized speeds
   (6) Variables    Qa, Qb, Qc      % Angles for A, B, C
   (7) Constants    LA, LB, LC      % Lengths of A, B, C
   (8) Specified    Vx, Vy          % Specified motion of P
   (9) Simprot(N, A, 3, QA)
-> (10) N_A = [COS(QA), -SIN(QA), 0; SIN(QA), COS(QA), 0; 0, 0, 1]
   (11) Simprot(N, B, 3, QB)
-> (12) N_B = [COS(QB), -SIN(QB), 0; SIN(QB), COS(QB), 0; 0, 0, 1]
   (13) Simprot(N, C, 3, QC)
-> (14) N_C = [COS(QC), -SIN(QC), 0; SIN(QC), COS(QC), 0; 0, 0, 1]
   (15) V_P_N> = LA*U1*A2> + LB*U2*B2> + LC*U3*C2>
-> (16) V_P_N> = LA*U1*A2> + LB*U2*B2> + LC*U3*C2>
   (17) DEPENDENT[1] = DOT(V_P_N>,N1>) - Vx
-> (18) DEPENDENT[1] = -VX - LA*SIN(QA)*U1 - LB*SIN(QB)*U2 - LC*SIN(QC)*U3
   (19) DEPENDENT[2] = DOT(V_P_N>,N2>) - Vy
-> (20) DEPENDENT[2] = LA*COS(QA)*U1 + LB*COS(QB)*U2 + LC*COS(QC)*U3 - VY
   (21) Constrain( DEPENDENT[U2,U3] )
-> (22) U2 = -(VX*COS(QC)+VY*SIN(QC)+LA*SIN(QA-QC)*U1)/(LB*SIN(QB-QC))
-> (23) U3 = (VX*COS(QB)+VY*SIN(QB)+LA*SIN(QA-QB)*U1)/(LC*SIN(QB-QC))
```

5 Example Problems

5.1 Mathematical Capabilities Examples

```
   (1) %     File:  tutor1.al    [Autolev: http://www.autolev.com]
   (2) %  Problem:  Examples of mathematical functions
   (3) %--------------------------------------------------------------------
   (4) %       Mathematical declarations: variables, constants
   (5) Variables  X', Y'
   (6) Constants  A,B,C,D
   (7) %--------------------------------------------------------------------
   (8) %       Create an expression and then rearrange (simplify) it
   (9) E = (X+2*Y)^2 + 3*(7+X)*(X+Y)         % Create an expression
-> (10) E = (X+2*Y)^2 + 3*(7+X)*(X+Y)

   (11) Expand( E, 1:2 )                      % Clear parentheses
-> (12) E = 21*X + 21*Y + 4*X^2 + 4*Y^2 + 7*X*Y

   (13) Factor( E, X )                        % Factor on X
-> (14) E = 21*Y + 4*Y^2 + 4*X*(5.25+X+1.75*Y)

   (15) %--------------------------------------------------------------------
   (16) %       Mathematical commands
   (17) Dy = D( E, Y )                        % Partial derivative wrt. Y
-> (18) DY = 21 + 7*X + 8*Y

   (19) Dt = Dt( E )                          % Total derivative wrt. t
-> (20) DT = 21*Y' + 8*Y*Y' + 4*X*(X'+1.75*Y') + 4*(5.25+X+1.75*Y)*X'

   (21) TY = Taylor( X*COS(Y), 0:7, X=0,Y=0)  % Multi-variable Taylor series
-> (22) TY = 0.001388889*X*(720+30*Y^4-360*Y^2-Y^6)

   (23) F = Evaluate(TY, X=1, Y=0.5)          % Symbolic/numerical evaluation
-> (24) F = 0.8775825

   (25) POLY = Polynomial( [A,B,C], X )       % Creates A*X^2 +B*X +C
-> (26) POLY = C + B*X + A*X^2
```

```
   (27) ROOT1 = Roots( [1; 2; 3; 4] )         % Roots of X^3+2*X^2+3*X+4 = 0
-> (28) ROOT1 = [-1.650629; -0.1746854 - 1.546869*i; -0.1746854 + 1.546869*i]

   (29) ROOT2 = Roots( POLY, X, 2 )           % Some symbolic roots
-> (30) ROOT2[1] = -0.5*(B-(B^2-4*A*C)^0.5)/A
-> (31) ROOT2[2] = -0.5*(B+(B^2-4*A*C)^0.5)/A

   (32) %-------------------------------------------------------------------
   (33) %       Creating row or column matrices
   (34) RowMatrix = [1, 2, 3, 4]     % Create a 1x4 matrix
-> (35) ROWMATRIX = [1, 2, 3, 4]

   (36) ColMatrix = [1; 2; 3; 4]     % Create a 4x1 matrix
-> (37) COLMATRIX = [1; 2; 3; 4]

   (38) Zero[1] = A*X + B*Y - 1      % Assign elements of column matrix
-> (39) ZERO[1] = -1 + A*X + B*Y

   (40) Zero[2] = C*X + D*Y - Pi
-> (41) ZERO[2] = -3.141593 + C*X + D*Y

   (42) %-------------------------------------------------------------------
   (43) %       Solve a set of linear equations
   (44) Solve( Zero, X, Y )
-> (45) X = -(3.141593*B-D)/(A*D-B*C)
-> (46) Y = (3.141593*A-C)/(A*D-B*C)

   (47) %-------------------------------------------------------------------
   (48) %       Creating rectangular matrices
   (49) M0 = [A, B; C, 0]            % Create a 2x2 matrix
-> (50) M0 = [A, B; C, 0]

   (51) M0[2,2] := D                 % Assign element of rectangular matrix
-> (52) M0[2,2] = D

   (53) M1 = [M0, [1,2;3,4] ]        % Elements of matrices can be matrices
-> (54) M1 = [A, B, 1, 2; C, D, 3, 4]

   (55) %-------------------------------------------------------------------
   (56) %       Matrix commands
   (57) M2 = M0 + M0                 % Matrix addition
-> (58) M2 = [2*A, 2*B; 2*C, 2*D]

   (59) M3 = M0 * M0                 % Matrix multiplication
-> (60) M3 = [A^2 + B*C, B*(A+D); C*(A+D), B*C + D^2]

   (61) M4 = Transpose(M0)           % Matrix transposition
```

```
-> (62) M4 = [A, C; B, D]

   (63) M5 = Inv(M0)                % Matrix inversion
-> (64) M5[1,1] = D/(A*D-B*C)
-> (65) M5[1,2] = -B/(A*D-B*C)
-> (66) M5[2,1] = -C/(A*D-B*C)
-> (67) M5[2,2] = A/(A*D-B*C)

   (68) M6 = Diagmat(3,1)           % Makes 3x3 matrix with 1 along diagonal
-> (69) M6 = [1, 0, 0; 0, 1, 0; 0, 0, 1]

   (70) M7 = Cols( M0, 1 )          % Returns column 1 of M0
-> (71) M7 = [A; C]

   (72) M8 = Rows( M0, 2, 1:2)      % Returns rows 2 and rows 1 through 2 of M0
-> (73) M8 = [C, D; A, B; C, D]

   (74) N1 = Rows( M0 )             % Returns the number of rows in M0
-> (75) N1 = 2

   (76) N2 = Det( M0 )              % Determinant of a matrix
-> (77) N2 = A*D - B*C

   (78) M9 = Evaluate( M0, A=1, B=2, C=3, D=4 )
-> (79) M9 = [1, 2; 3, 4]

   (80) LAMBDA = Eig( M9 )          % Eigenvalues
-> (81) LAMBDA = [-0.3722813; 5.372281]

   (82) Eig( M9, EigValue, EigVec)  % Eigenvalues/eigenvectors
-> (83) EIGVALUE = [-0.3722813; 5.372281]
-> (84) EIGVEC = [-0.8245648, 0.5657675; -0.4159736, -0.9093767]

   (85) %---------------------------------------------------------------------
   (86) %       Record Autolev responses
```

5.2 Mass, Mass Center, and Inertia Calculations



The schematic above shows three identical uniform hinge-connected plates A, B, and C. Dextral (right-handed) sets of mutually perpendicular unit vectors Ai>, Bi>, and Ci>, (i=1,2,3), are fixed in A, B, and C, as shown in the figure above, with A2> = B2> = C2> all parallel to the axes of the hinge connecting A and B. The plates are thin and square and have dimension W=1 meter and are of mass M=12 Kg. Points O and P mark corners of A and C and the angle PHI characterizes the orientation of B in A, as shown above.

- Find an expression for the distance between point O and the center of mass of the system formed by A, B, and C.
  Result:
  DISTANCE = (0.94444-0.22222*COS(PHI)-0.055556*SIN(PHI))^0.5
- For PHI = 90°, find the central principal moments of inertia of the system and the angle between A2> and the principal axes associated with the minimum moment of inertia of the system.
  Result:
  LAMBDA = [5; 13; 14] ANGLE = 65.90516°
- Plot the distance from point O to the center of mass of the system vs. PHI for 0 ≤ PHI ≤ 360°. Determine the value of PHI for which this distance is minimum, and find this distance. On the same plot, plot the radius of gyration of the system about the line joining points O and P. Determine the value of PHI for which the radius of gyration is a minimum, and find this value.
  Result: See the plot below.
  PHI = 14 degrees DISTANCE = 0.84580 meters

```
PHI = 340 degrees       GYRATION = 0.696   meters
```



```
%     File:  tutor2.al    [Autolev: http://www.autolev.com]
%  Problem:  Mass and inertia calculations
%--------------------------------------------------------------------
Degrees    ON               % Online calculations of PHI in degrees
%--------------------------------------------------------------------
%       Physical declarations: bodies, points
Bodies     A, B, C          % Right, middle, top plate
Points     O, P             % Points on L
Points     CM               % Center of mass of system
%--------------------------------------------------------------------
%       Mathematical declarations: variables, constants
Variables  PHI              % Angle between plates A and B
Constants  W=1              % Width of plate
%--------------------------------------------------------------------
%       Mass and inertia
Mass       A=M=12, B=M, C=M
Inertia    A, M*W^2/12, M*W^2/12, M*W^2/6
Inertia    B, M*W^2/12, M*W^2/12, M*W^2/6
Inertia    C, M*W^2/12, M*W^2/6,  M*W^2/12
%--------------------------------------------------------------------
%       Geometry relating unit vectors
Simprot(A, B, -2, PHI)
A_C = Diagmat(3,1)          % 3x3 identity matrix
%--------------------------------------------------------------------
%       Position vectors     (Ao, Bo, Co are mass centers)
P_O_Ao> = -0.5*W*A1> + 0.5*W*A2>
P_O_Bo> = -W*A1> + 0.5*W*A2> + 0.5*W*B1>
P_O_Co> = W*A2> - 0.5*W*C1> - 0.5*W*C3>
P_O_P> =  W*A2> - W*C3>
%--------------------------------------------------------------------
%       Center of mass of system
P_O_CM> = CM(O)
Distance = Mag( P_O_CM> )
%--------------------------------------------------------------------
%       Inertia dyadics of system about CM and O
I_SYSTEM_CM>> = Inertia(CM, A, B, C)        % Central inertia dyadic
I_SYSTEM_CM = Represent( I_SYSTEM_CM>>, A)  % Central inertia matrix
Eig( Evaluate(I_SYSTEM_CM, PHI=90), LAMBDA, EIGENVECS )
EVEC = Rows(EIGENVECS, 1)                   % Extract eigenvector
EVEC> = Represent( EVEC, A )                % Corresponding unit vector
ANGLE = acos( DOT(EVEC>,A2>) )              % Angle between EVEC>, A2>
%--------------------------------------------------------------------
%       Moment of inertia of system about line joining O and P
I_SYSTEM_O>> = Inertia(O)              % Inertia dyadic about point O
Express(I_SYSTEM_O>>, A)               % Express in terms of A1>, A2>, A3>
L> = Unitvec( P_O_P> )                 % Unit vector parallel to line L
IL = Dot(L>, DOT(I_SYSTEM_O>>, L>) )   % Moment of inertia about L
MTOTAL = Mass()                        % Mass of system
GYRATION = (IL / MTOTAL)^0.5           % Radius of gyration about L
%--------------------------------------------------------------------
%       Output and units for use in program created by CODE command
UnitSystem  kg, meter, sec
Output PHI degrees, DISTANCE meters, GYRATION meters
%--------------------------------------------------------------------
%       C code generation for numerical solution
CODE Algebraic() [PHI deg=0,360,1] tutor2.c
%--------------------------------------------------------------------
%       Record Autolev responses
Save tutor2.all
```

5.3 Solving a Set of Nonlinear Equations



The four-bar linkage pictured above consists of links `A`, `B`, `C`, and ground `N`. Link `A` is connected by means of revolute joints to `N` and `B` at points `O` and `P`, respectively. Link `C` is connected with revolute joints to `N` and `B` at points `R` and `Q`, respectively. Dextral (right-handed) sets of mutually perpendicular unit vectors `Ni>`, `Ai>`, `Bi>`, and `Ci>`, (i=1,2,3), are fixed in `N`, `A`, `B`, and `C`, as shown above, with `N1>` vertically downward, `A1>` directed from `O` to `P`, `B1>` directed from `P` to `Q`, `C1>` directed from `R` to `Q`, and `N3>` = `A3>` = `B3>` = `C3>` all parallel to the axes of the revolute joints. The distances from `O` to `P`, `P` to `Q`, `Q` to `R`, and `R` to `O` are `LA`, `LB`, `LC`, and `LN`, respectively. The angles `QA`, `QB`, `QC` characterize the orientation of `A`, `B`, and `C` in `N`, as shown.

- Create a pair of configuration constraint equations which relate `QA`, `QB`, and `QC`.
  Result:
  ```
  CONFIG[1] = LA*COS(QA) + LB*COS(QB) - LC*COS(QC)
  CONFIG[2] = LA*SIN(QA) + LB*SIN(QB) - LC*SIN(QC) - LN
  ```
- Determine the values of `QB` and `QC` which satisfy the configuration constraint equations when `LA=LN=1 m, LB=LC=2 m, QA=30°`.
  Result:
  `QB` = 74.4775° `QC` = 45.5225°

```
%     File:  tutor3.al    [Autolev: http://www.autolev.com]
%  Problem:  Solving a set of nonlinear equations (4-bar linkage)
%----------------------------------------------------------------------
%       Physical declarations: Frames and Points
Frames     A, B, C, N
Points     O, P, Q, R
%----------------------------------------------------------------------
%       Mathematical declarations:  variables, constants
Variables  QA, QB, QC        % Configuration variables
Constants  LA, LB, LC, LN    % Lengths
%----------------------------------------------------------------------
%       Geometry relating unit vectors
Simprot(N, A, 3, QA)
Simprot(N, B, 3, QB)
Simprot(N, C, 3, QC)
%----------------------------------------------------------------------
%       Position vectors
P_O_P> = LA*A1>
P_P_Q> = LB*B1>
P_R_Q> = LC*C1>
P_O_R> = LN*N2>
%----------------------------------------------------------------------
%       Configuration constraints
LOOP> = P_O_P> + P_P_Q> + P_Q_R> + P_R_O>
CONFIG[1] = Dot(LOOP>,N1>)
CONFIG[2] = Dot(LOOP>,N2>)
%----------------------------------------------------------------------
%       Input constants, variables, etc. for Fortran code
UnitSystem  kg, meter, sec
Input  LA=1 m,  LB=2 m,  LC=2 m,  LN=1 m,  QA=30 deg
Input  QB=60 deg, QC=20 deg            % Guess
Input  absErr=1.0E-07, relErr=1.0E-07  % Error tolerances
%----------------------------------------------------------------------
%       Fortran code generation for numerical solution
CODE Nonlinear(CONFIG,QB,QC) tutor3.for
%----------------------------------------------------------------------
% Comment:  Nonlinear equations may have more than one solution.
%           The solution depends on a guess.  If a guess does not
%           provide a satisfactory answer, try another guess.
%----------------------------------------------------------------------
%       Record Autolev responses
Save tutor3.all
```

5.4 Spring-Restrained Double Pendulum – Equilibrium Configuration



The spring-restrained double pendulum depicted above consists of two light rigid rods, A and B, and a linear spring that supports two particles P and Q in a Newtonian reference frame N. Rod A is connected with frictionless revolute joints to N and B at points O and P, respectively. Dextral sets of mutually perpendicular unit vectors Ni>, Ai>, and Bi>, (i=1,2,3), are fixed in N, A, and B, as shown, with N1> vertically downward, A1> directed from O to P, B1> directed from P to Q, and N3> = A3> = B3> all parallel to the axes of the revolute joints. The distance from O to P is LA, and the distance from P to Q is LB. The masses of P and Q are MP and MQ, respectively. The spring's natural length is Lo, and its spring constant is K. The angles QA and QB characterize the orientation of A and B in N, as shown, and the generalized speeds U1 and U2 are defined as U1 = LA*QA', U2 = LB*QB'.

- Formulate a set of equations that govern the static equilibrium of the system.

  Result:

```
LOQ = (LA^2+LB^2+2*LA*LB*COS(QA-QB))^0.5
STRETCH = LOQ - LO
ZERO[1] = K*LB*SIN(QA-QB)*STRETCH/LOQ - G*(MP+MQ)*SIN(QA)
ZERO[2] = -G*MQ*SIN(QB) - K*LA*SIN(QA-QB)*STRETCH/LOQ
```

- Determine three pairs of values for QA and QB which satisfy the static equilibrium equations when LA=1 m, LB=2 m, MP=10 kg, MQ=20 kg, Lo=1 m, K=200 n/m, G=9.81 m/sec^2.

  Result with a guess of QA=$-30°$ and QB=30°:

  QA = $-50.1492°$ QB = $35.1548°$

  Result with a guess of QA= $30°$ and QB=$-30°$:

  QA = $50.1492°$ QB = $-35.1548°$

  Result with a guess of QA=20° and QB=60°:

  QA = $0.0°$ QB = $0.0°$

```
%     File:  tutor4.al    [Autolev: http://www.autolev.com]
%  Problem:  Spring-Restrained Double Pendulum -- Equilibrium Configuration
%----------------------------------------------------------------------
%       Newtonian, bodies, frames, particles, points
Newtonian    N
Frames       A, B
Particles    P, Q
Points       O
%----------------------------------------------------------------------
%       Mathematical declarations: variables, constants, mass
Variables    U{2}          % Generalized speeds
Variables    QA', QB'      % Generalized coordinates; derivatives
Constants    G, LA, LB     % Gravitational acceleration, lengths of A, B
Constants    K, LO         % Spring constant, natural length of spring
Mass         P=MP, Q=MQ
%----------------------------------------------------------------------
%       Geometry relating unit vectors
Simprot(N, A, 3, QA)
Simprot(N, B, 3, QB)
%----------------------------------------------------------------------
%       Position vectors
P_O_P> = LA*A1>
P_P_Q> = LB*B1>
%----------------------------------------------------------------------
%       Kinematical differential equations
QA' = U1/LA
QB' = U2/LB
%----------------------------------------------------------------------
%       Angular velocities
W_A_N> = QA'*A3>
W_B_N> = QB'*B3>
%----------------------------------------------------------------------
%       Velocities
V_O_N> = 0>
V2pts(N, A, O, P)
V2pts(N, B, P, Q)
%----------------------------------------------------------------------
%       Forces
Gravity( G*N1> )
LOQ = MAG(P_O_Q>)                    % Distance from O to Q
STRETCH = LOQ - LO                   % Stretch of spring
UVEC> = P_O_Q> / LOQ                 % Unit vector from O to Q
Force(O/Q, -K*STRETCH*UVEC> )        % Spring force
%----------------------------------------------------------------------
%       Equations of motion
Zero = Fr()
Factor(Zero, G)                      % Simplify slightly
%----------------------------------------------------------------------
%       Units system for CODE input/output conversions
UnitSystem  kg,meter,sec
%----------------------------------------------------------------------
%       Input constants, variables, etc. for CODE
Input  G=9.81 m/sec^2, LA=1 m, LB=2 m,  LO=1 m, K=200 n/m,  MP=10 kg, MQ=20 kg
Input  QA=-30 deg,  QB=30 deg        % Guess
Input  absErr=1.0E-07, relErr=1.0E-07  % Error tolerances
%----------------------------------------------------------------------
%       C code generation for numerical solution
CODE Nonlinear(Zero,QA,QB) tutor4.c
%----------------------------------------------------------------------
%       Record Autolev responses
Save tutor4.all
```

5.5 Four-Bar Linkage – Equilibrium Configuration



The four-bar linkage pictured above consists of links A, B, C, and ground N. Link A is connected by means of frictionless revolute joints to N and B at points O and P, respectively. Link C is connected with frictionless revolute joints to N and B at points R and Q, respectively. Dextral sets of mutually perpendicular unit vectors Ni>, Ai>, Bi>, and Ci> (i=1,2,3) are fixed in N, A, B, and C, as shown above, with N1> vertically downward, A1> directed from O to P, B1> directed from P to Q, C1> directed from R to Q, and N3> = A3> = B3> = C3> all parallel to the axes of the revolute joints. The distances from O to P, P to Q, Q to R, and R to O are LA, LB, LC, and LN, respectively. The masses of A, B, and C are MA, MB, and MC. A horizontal force H*N2> is applied to Q. The angles QA, QB, and QC characterize the orientation of A, B, and C in N, as shown, and the generalized speeds U1, U2, and U3 are defined as U1 = QA', U2 = QB', and U3 = QC'.

- For LA=LN=1 m, LB=LC=2 m, MA=10 kg, MB=20 kg, MC=20 kg, G=9.81 m/sec^2, formulate a set of equations governing the static equilibrium of the system.
  Result:

```
ZERO[1] = H*(COS(QA)-COS(QB)*SIN(QA-QC)/SIN(QB-QC)) + 98.1*(3*SIN(QB)*
          SIN(QA-QC)-SIN(QC)*SIN(QA-QB))/SIN(QB-QC) - 441.45*SIN(QA)
ZERO[2] = COS(QA) + 2*COS(QB) - 2*COS(QC)
ZERO[3] = SIN(QA) + 2*SIN(QB) - 2*SIN(QC) - 1
```

- Determine the values of QA, QB, QC which satisfy the equilibrium equations when H=200 N:
  Result:

  QA = 19.987° QB = 71.662° QC = 38.325°

```
%     File:  tutor5.al    [Autolev: http://www.autolev.com]
%  Problem:  Four-Bar Linkage -- Equilibrium Configuration
%----------------------------------------------------------------
%       Newtonian, bodies, frames, particles, points
Newtonian    N
Bodies       A, B, C
Points       O, P, Q, R
%----------------------------------------------------------------
%       Variables, constants, and specified
Variables    U{3}                    % Generalized speeds
Variables    QA', QB', QC'           % Generalized coordinates; derivatives
Constants    G=9.81                  % Gravitational acceleration
Constants    LA=1, LB=2, LC=2, LN=1 % Lengths
Constants    H                       % Horizontal force on Q
%----------------------------------------------------------------
%       Mass and inertia
Mass         A=MA=10, B=MB=20, C=MC=20
Inertia      A, 0, MA*LA^2/12, MA*LA^2/12
Inertia      B, 0, MB*LB^2/12, MB*LB^2/12
Inertia      C, 0, MC*LC^2/12, MC*LC^2/12
%----------------------------------------------------------------
%       Geometry relating unit vectors
Simprot(N, A, 3, QA)
Simprot(N, B, 3, QB)
Simprot(N, C, 3, QC)
%----------------------------------------------------------------
%       Position vectors
P_O_P> = LA*A1>
P_P_Q> = LB*B1>
P_R_Q> = LC*C1>
P_O_R> = LN*N2>
P_O_Ao> = 0.5*P_O_P>
P_P_Bo> = 0.5*P_P_Q>
P_R_Co> = 0.5*P_R_Q>
%----------------------------------------------------------------
%       Kinematical differential equations
QA' = U1
QB' = U2
QC' = U3
%----------------------------------------------------------------
%       Angular velocities
W_A_N> = QA'*A3>
W_B_N> = QB'*B3>
W_C_N> = QC'*C3>
%----------------------------------------------------------------
%       Configuration Constraints
LOOP> = P_O_P> + P_P_Q> + P_Q_R> + P_R_O>
```

```
CONFIG[1] = Dot(LOOP>,N1>)
CONFIG[2] = Dot(LOOP>,N2>)
%----------------------------------------------------------------------
%       Velocities
V_O_N> = 0>
V2pts(N, A, O, Ao)
V2pts(N, A, O, P)
V2pts(N, B, P, Bo)
V2pts(N, B, P, Q)
V2pts(N, C, Q, Co)
V2pts(N, C, Q, R)
%----------------------------------------------------------------------
%       Motion constraints
DEPENDENT[1] = Dot(V_R_N>, N1>)
DEPENDENT[2] = Dot(V_R_N>, N2>)
Constrain( DEPENDENT[U2,U3] )
%----------------------------------------------------------------------
%       Forces
Gravity( G*N1> )
Force_Q> = H*N2>
%----------------------------------------------------------------------
%       Equations of motion
Zero = Fr()
Zero[2] = CONFIG[1]
Zero[3] = CONFIG[2]
%----------------------------------------------------------------------
%       Units system for CODE input/output conversions
UnitSystem  kg,meter,sec
%----------------------------------------------------------------------
%       Input constants, variables, etc. for CODE
Input  H=200 newton
Input  QA=30 deg, QB=60 deg, QC=30 deg   % Guess
Input  absErr=1.0E-07, relErr=1.0E-07    % Error tolerances
%----------------------------------------------------------------------
%       MATLAB code generation for numerical solution
CODE Nonlinear( Zero, QA,QB,QC )  tutor5.m
%----------------------------------------------------------------------
%       Record Autolev responses
Save tutor5.all
```

5.6 Spring-Damper-Restrained Double Pendulum – Motion and Contact Forces



The spring-damper-restrained double pendulum depicted above consists of two light rigid rods A and B, and a linear spring-damper that supports two particles P and Q in a Newtonian reference frame N. Rod A is connected with frictionless revolute joints to N and B at points O and P, respectively. Dextral sets of mutually perpendicular unit vectors Ni>, Ai>, and Bi>, (i=1,2,3), are fixed in N, A, and B, as shown in the figure, with N1> vertically downward, A1> directed from O to P, B1> directed from P to Q, and N3> = A3> = B3> all perpendicular to the plane in which A and B move. The distance from O to P is LA, and the distance from P to Q is LB. The masses of P and Q are MP and MQ, respectively. The spring's natural length is Lo, its spring constant is K, and the damping constant is C. The angles QA and QB characterize the orientation of A and B in N, as shown, and the generalized speeds U1 and U2 are defined as U1 = LA*QA' and U2 = LB*QB'.

The set of contact forces exerted by N on A across the revolute joint at O is equivalent to a couple of torque T1*N1>+T2*N2> together with a force F1*N1>+F2*N2>+F3*N3> applied at O. A damping force Fp*A2> is applied to P, where Fp = -D*QA', D being a constant.

- Find the equations which govern the motion of the system.

  Result:

```
LOQ = (LA^2+LB^2+2*LA*LB*COS(QA-QB))^0.5
STRETCH = LOQ - LO
STRETCH' = SIN(QA-QB)*(LA*U2-LB*U1)/LOQ
ZERO[1] = FP + SIN(QA-QB)*(LB^2*(K*STRETCH+C*STRETCH')/LOQ-MQ*U2^2)/LB
        - G*(MP+MQ)*SIN(QA) - (MP+MQ)*U1' - MQ*COS(QA-QB)*U2'
```

```
ZERO[2] = MQ*SIN(QA-QB)*U1^2/LA - G*MQ*SIN(QB) - MQ*U2' - MQ*COS(QA-QB)*U1'
        - LA*(K*STRETCH+C*STRETCH')*SIN(QA-QB)/LOQ
```

- Plot the time history of STRETCH for $0 \leq t \leq 10$ sec, and determine the values of QA and QB when the system comes to rest. Form a numerical integration checking function and verify that it remains constant during numerical integration. The system has the parameters LA=1 m, LB=2 m, MA=10 kg, MB=20 kg, Lo=1 m, K=200 n/m, C=300 n*sec/m, D=300 n*sec, G=9.81 m/sec^2, and the initial values QA=20°, QB=60°, U1=U2=0.

  Results:

  See the plot of the time-history of STRETCH.

  The file tutor6.1 contains the time-history of ECHECK, which remains constant.

  The values of QA and QB when the system comes to rest are QA = $-50.15°$ and QB = $35.15°$.

- Find expressions for F1 and F2 and plot these quantities for $0 \leq t \leq 10$ sec.

  Note: To find F1 and F2, modify the file tutor6.al as follows.

  Change the declaration of generalized speeds to `Variables U{4}'`

  Change the velocity of point O in N to `V_O_N> = U3*N1> + U4*N2>`

  Uncomment the lines

  ```
  AUXILIARY[1] = DOT(V_O_N>, N1>)
  AUXILIARY[2] = DOT(V_O_N>, N2>)
  Constrain( AUXILIARY[U3,U4] )
  ```

  Change the KANE command to `Kane( F1, F2 )`

  Uncomment the line `Output T, F1, F2`

  Results:

  See the plot of the time-histories of F1 and F2.

```
F1 = FP*SIN(QA) - G*(MP+MQ) - MP*COS(QA)*U1^2/LA - MQ*SIN(QB)*U2'
   - (MP+MQ)*SIN(QA)*U1' - MQ*(COS(QA)*U1^2/LA+COS(QB)*U2^2/LB)
F2 = MQ*COS(QB)*U2' + (MP+MQ)*COS(QA)*U1' - FP*COS(QA) - MP*SIN(QA)*U1^2/LA
   - MQ*(SIN(QA)*U1^2/LA+SIN(QB)*U2^2/LB)
```


Time History of Stretch
STRETCH (meters)
2
1.9
1.8
1.7
1.6
1.5
1.4
1.3
0
5
10
15
20
Time (seconds)


Time Histories of F1 and F2
F1, F2 (newtons)
200
100
0
-100
-200
-300
-400
F1 (NEWTONS)
F2 (NEWTONS)
0
5
10
15
20
TIME (seconds)

```
%     File:  tutor6.al    [Autolev: http://www.autolev.com]
%  Problem:  Spring-Damper-Restrained Double Pendulum --
%            Motion and Contact Forces
%----------------------------------------------------------------------
%       Newtonian, bodies, frames, particles, points
Newtonian    N
Frames       A, B
Particles    P, Q
Points       O
%----------------------------------------------------------------------
%       Variables, constants, and specified
Variables    U{2}'           % Generalized Speeds; derivatives
Variables    QA', QB'        % Generalized coordinates; derivatives
Variables    F1, F2          % Contact forces
Constants    G, LA, LB       % Gravitational acceleration, lengths of A, B
Constants    K, LO           % Spring constant, natural length of spring
Constants    C, D            % Damping constants
Specified    STRETCH'        % Elongation of spring joining O and Q
Specified    FP = -D*QA'     % Force on P
%----------------------------------------------------------------------
%       Mass and inertia
Mass         P=MP, Q=MQ
%----------------------------------------------------------------------
%       Geometry relating unit vectors
Simprot(N, A, 3, QA)
Simprot(N, B, 3, QB)
%----------------------------------------------------------------------
%       Position vectors
P_O_P> = LA*A1>
P_P_Q> = LB*B1>
%----------------------------------------------------------------------
%       Kinematical differential equations
QA' = U1/LA
QB' = U2/LB
%----------------------------------------------------------------------
%       Angular velocities
W_A_N> = QA'*A3>
W_B_N> = QB'*B3>
%----------------------------------------------------------------------
%       Velocities
% V_O_N> = U3*N1> + U4*N2>            % Brings into evidence F1, F2
V_O_N> = 0>
V2pts(N, A, O, P)
V2pts(N, B, P, Q)
%----------------------------------------------------------------------
%       Motion constraints
% AUXILIARY[1] = Dot(V_O_N>, N1>)      % Dot(V_O_N>,N1>) = 0
```

```
% AUXILIARY[2] = Dot(V_O_N>, N2>)      % Dot(V_O_N>,N2>) = 0
% Constrain( AUXILIARY[U3,U4] )       % Solves for U3,U4
%--------------------------------------------------------------------
%       Forces
Gravity( G*N1> )
Force_P> += Fp*A2>
LOQ = Mag(P_O_Q>)                     % Distance from O to Q
STRETCH = LOQ - LO                    % Stretch of spring
UVEC> = P_O_Q> / LOQ                  % Unit vector directed from O to Q
STRETCH' = Dot(V_Q_N>,UVEC>)          % Time rate of change of stretch
Force(O/Q, (-K*STRETCH-C*STRETCH')*UVEC> )
Force_O> += F1*N1> + F2*N2>
%--------------------------------------------------------------------
%       Equations of motion
Zero = Fr() + FrStar()
Factor(Zero, G, SIN(QA-QB) )          % Simplify slightly
Kane()
% Kane(F1,F2)                         % Solve for F1, F2
%--------------------------------------------------------------------
%       Units system for CODE input/output conversions
UnitSystem  kg,meter,sec
%--------------------------------------------------------------------
%       Input constants, variables, etc. for CODE
Input  G=9.81 m/sec^2, LA=1 m,  LB=2 m
Input  LO=1 m,  K=200 n/m
Input  C=300 n*sec/m, D=300 n*sec/m
Input  MP=10 kg,  MQ=20 kg
Input  QA=20 deg, QB=60 deg           % Initial values
Input  U1=0 rad/sec, U2=0  rad/sec    % Initial values
Input  tInitial=0,     tFinal=20      % Begin/end times
Input  integStp=0.2,   printInt=1     % Integration/print step
Input  absErr=1.0E-07, relErr=1.0E-07  % Error tolerances
%--------------------------------------------------------------------
%       Quantities to be output from CODE
ECHECK = NiCheck()                    % Checking function
Output T, STRETCH m, QA deg, QB deg, ECHECK joules
% Output T, F1 newtons, F2 newtons
%--------------------------------------------------------------------
%       C code generation for numerical solution
CODE Dynamics() tutor6.c
%--------------------------------------------------------------------
%       Record Autolev responses
Save tutor6.all
```

5.7 Four-Bar Linkage – Motion and Contact Forces



The four-bar linkage pictured above consists of links A, B, C, and ground N. Link A is connected by means of frictionless revolute joints to N and B at points O and P, respectively. Link C is connected with frictionless revolute joints to N and B at points R and Q, respectively. Dextral sets of mutually perpendicular unit vectors Ni>, Ai>, Bi>, and Ci> (i=1,2,3) are fixed in N, A, B, and C, as shown above, with N1> vertically downward, A1> directed from O to P, B1> directed from P to Q, C1> directed from R to Q, and N3> = A3> = B3> = C3> all perpendicular to the plane in which A, B, and C move. The distances from O to P, P to Q, Q to R, and R to O are LA, LB, LC, and LN, respectively. The masses of A, B, and C are MA, MB, and MC. The angles QA, QB, QC characterize the orientation of A, B, and C in N, as shown, and the generalized speeds U1, U2, and U3 are defined as U1 = QA', U2 = QB', and U3 = QC'.

The set of contact forces exerted by N on A across the revolute joint at O is equivalent to a couple of torque TA1*N1>+TA2*N2> together with a force FA1*N1>+FA2*N2>+FA3*N3> applied at O. Similarly, the set of contact forces exerted by N on C across the revolute joint at R is equivalent to a couple of torque TC1*N1>+TC2*N2> together with a force FC1*N1>+FC2*N2>+FC3*N3> applied at R.

- For LA=LN=1 m, LB=LC=2 m, MA=10 kg, MB=20 kg, MC=20 kg, and the initial values QA=30° and U1=0, plot the angles QA, QB, QC for 0 $\leq$ t $\leq$ 10 sec. Note: The values QB=74.47751219° and QC=45.52248781° satisfy the configuration constraint equations. Form a numerical integration checking function and verify that it remains constant during numerical integration. Verify that the configuration constraint equations, which also serve as numerical integration checking functions, are satisfied during numerical integration.

  Results:

See the plot of the time-histories of QA, QB, and QC.

See file tutor7.2 for the time-histories of CONFIG[1], CONFIG[2], and ECHECK.

- Plot the time histories of FC1 and FC2 for 0 $\leq$ t $\leq$ 10 sec.

  Note: To find FC1 and FC2, modify the file tutor7.al as follows.

  Change the lines

```
DEPENDENT[1] = DOT(V_R_N>, N1>)
DEPENDENT[2] = DOT(V_R_N>, N2>)
Constrain( DEPENDENT[U2,U3] )
Kane()
```

  to

```
AUXILIARY[1] = DOT(V_R_N>, N1>)
AUXILIARY[2] = DOT(V_R_N>, N2>)
Constrain( AUXILIARY[U2,U3] )
Kane(FC1,FC2)
```

  Uncomment the lines

```
Force_R> = FC1*N1> + FC2*N2>
Output T, FC1, FC2
```

  Result:

  See the plot of the time-histories of FC1 and FC2.

- Plot the time-history of FA1 for 0 $\leq$ t $\leq$ 10 sec.

  Note: To find FA1, modify the file tutor7.al once more.

  Change the declaration of generalized speeds to `Variables U{4}'`

  Change the velocity of point O in N to `V_O_N> = U4*N1>`

  Uncomment the line `AUXILIARY[3] = DOT(V_O_N>, N1>)`

  Change the Constrain command to `Constrain( AUXILIARY[U2,U3,U4] )`

  Change the KANE command to `Kane( FC1,FC2,FA1 )`

  Uncomment the line `Output T, FA1`

  Result:

  See the plot of the time-history of FA1.


Time Histories of QA, QB, QC
QA, QB, QC (degrees)
Time (seconds)
QA (DEG)
QB (DEG)
QC (DEG)
80
60
40
20
0
-20
-40
-60
-80
0
2
4
6
8
10


Time Histories of FC1 and FC2
FC1, FC2 (newtons)
Time (seconds)
FC1
FC2
100
0
-100
-200
-300
-400
-500
0
2
4
6
8
10


Time History of FA1
FA1 (newtons)
Time (seconds)
-40
-60
-80
-100
-120
-140
-160
-180
-200
-220
-240
-260
0
2
4
6
8
10

```
%     File:  tutor7.al    [Autolev: http://www.autolev.com]
%  Problem:  Four-Bar Linkage -- Motion and Contact Forces
%----------------------------------------------------------------------
%       Default Settings
AutoZ        ON              % Program introduces Zees automatically
Digits       7               % Significant digits
%----------------------------------------------------------------------
%       Newtonian, bodies, frames, particles, points
Newtonian    N
Bodies       A, B, C
Points       O, P, Q, R
%----------------------------------------------------------------------
%       Variables, constants, and specified
Variables    U{3}'                      % Generalized speeds; derivatives
Variables    QA', QB', QC'              % Generalized coordinates; derivatives
Variables    FA{2}, FC{2}               % Contact forces
Constants    G=9.81                     % Gravitational acceleration
Constants    LA=1, LB=2, LC=2, LN=1     % Lengths
ZEE_NOT = [FA1, FA2, FC1, FC2]
%----------------------------------------------------------------------
%       Mass and inertia
Mass         A=MA=10, B=MB=20, C=MC=20
Inertia      A, 0, MA*LA^2/12, MA*LA^2/12
Inertia      B, 0, MB*LB^2/12, MB*LB^2/12
Inertia      C, 0, MC*LC^2/12, MC*LC^2/12
%----------------------------------------------------------------------
%       Geometry relating unit vectors
Simprot(N, A, 3, QA)
Simprot(N, B, 3, QB)
Simprot(N, C, 3, QC)
%----------------------------------------------------------------------
%       Position vectors
P_O_P> = LA*A1>
P_P_Q> = LB*B1>
P_R_Q> = LC*C1>
P_O_R> = LN*N2>
P_O_Ao> = 0.5*P_O_P>
P_P_Bo> = 0.5*P_P_Q>
P_R_Co> = 0.5*P_R_Q>
%----------------------------------------------------------------------
%       Kinematical differential equations
QA' = U1
QB' = U2
QC' = U3
%----------------------------------------------------------------------
%       Angular velocities
W_A_N> = QA'*A3>
W_B_N> = QB'*B3>
W_C_N> = QC'*C3>
%----------------------------------------------------------------------
%       Velocities
% V_O_N> = U4*N1>
V_O_N> = 0>
V2pts(N, A, O, Ao)
V2pts(N, A, O, P)
V2pts(N, B, P, Bo)
V2pts(N, B, P, Q)
V2pts(N, C, Q, Co)
V2pts(N, C, Q, R)
%----------------------------------------------------------------------
%       Motion constraints
```

```
DEPENDENT[1] = Dot(V_R_N>, N1>)             % Dot(V_R_N>, N1>) = 0
DEPENDENT[2] = Dot(V_R_N>, N2>)             % Dot(V_R_N>, N2>) = 0
Constrain( DEPENDENT[U2,U3] )              % Solve for U2, U3
% AUXILIARY[3] = Dot(V_O_N>, N1>)            % Dot(V_O_N>, N1>) = 0
% Constrain( AUXILIARY[U2,U3,U4] )           % Solve for U2,U3,U4
%----------------------------------------------------------------------
%       Angular accelerations
ALF_A_N> = DT(W_A_N>, N)
ALF_B_N> = DT(W_B_N>, N)
ALF_C_N> = DT(W_C_N>, N)
%----------------------------------------------------------------------
%       Accelerations of particles and mass centers of bodies
A_O_N> = 0>
A2pts(N, A, O, Ao)
A2pts(N, A, O, P)
A2pts(N, B, P, Bo)
A2pts(N, B, P, Q)
A2pts(N, C, Q, Co)
%----------------------------------------------------------------------
%       Forces
Gravity( G*N1> )
Force_O> = FA1*N1> + FA2*N2>
% Force_R> = FC1*N1> + FC2*N2>
%----------------------------------------------------------------------
%       Equations of motion
Zero = Fr() + FrStar()
Kane( )
% Kane( FC1, FC2, FA1 )
%----------------------------------------------------------------------
%       Units system for CODE input/output conversions
UnitSystem  kg,meter,sec
%----------------------------------------------------------------------
%       Input constants, variables, etc. for CODE
Input  QA=30 deg, QB=74.47751219 deg, QC=45.52248781 deg % Initial values
Input  tInitial=0,     tFinal=10                        % Begin/end times
Input  integStp=0.1,   printInt=1                       % Integration/print step
Input  absErr=1.0E-07, relErr=1.0E-07                   % Error tolerances
%----------------------------------------------------------------------
%       Quantities to be output from CODE
LOOP> = P_O_P> + P_P_Q> + P_Q_R> + P_R_O>
CONFIG[1] = Dot(LOOP>,N1>)                  % Should always equal 0
CONFIG[2] = Dot(LOOP>,N2>)                  % Should always equal 0
ECHECK = NICHECK()                          % Checking function
Output T, QA deg, QB deg, QC deg
Output T, CONFIG[1], CONFIG[2], ECHECK
% Output T, FC1 newtons, FC2 newtons
% Output T, FA1 newtons
%----------------------------------------------------------------------
%       Fortran code generation for numerical solution
CODE Dynamics() tutor7.for
%----------------------------------------------------------------------
%       Record Autolev responses
Save tutor7.all
```

5.8 Dynamics of a Cart Carrying an Inverted Pendulum



The sketch shows a thin uniform rod B attached to a cart A which slides on a smooth horizontal plane fixed in a Newtonian reference frame N. Right-handed sets of mutually perpendicular unit vectors Ni> and Bi> (i=1,2,3) are fixed in N and B, as shown, with N2> vertically upward, B2> parallel to the long axis of B, and N3> = B3> all parallel to the axis of the frictionless revolute joint which joins A and B at point Ao, the mass center of A. The length of B is L, and the masses of A and B are MA and MB, respectively. X is the distance from a point O fixed in N to Ao, and the angle THETA characterizes the orientation of B in N, as shown. The generalized speeds U1 and U2 are defined as U1 = X' and U2 = THETA'.

To control the cart, a force F*N1> is applied to Ao. A state-space controller is used so that F is specified in terms of constant feedback gains K_i (i=1,2,3,4) as F = K1*X + K2*THETA + K3*U1 + K4*U2.

- Form equations of motion for the system.
  Result: The output from the KANE command is

```
ZERO[1] = F + 0.5*L*MB*COS(THETA)*U2' - 0.5*L*MB*SIN(THETA)*U2^2 - (MA+MB)*U1'
ZERO[2] = 0.5*G*L*MB*SIN(THETA) + 0.5*L*MB*COS(THETA)*U1' - 0.25*(4*IB+MB*L^2)*U2'
```

- For L=1, MA=10, MB=1, G=9.81, show that X=THETA=U1=U2=U1'=U2'=0 is a solution of the equations of motion.
  Result: The output from CHECK = evaluate(ZERO, X=0, THETA=0, U1=0, U2=0, U1'=0, U2'=0) is CHECK = [0; 0]

- After introducing the variables dX, dTHETA, dU1, and dU2 as perturbations of X, THETA, U1, and U2, respectively, linearize the kinematical and dynamical equations in the perturbations about the nominal solution X=THETA=U1=U2=U1'=U2'=0. Put the linearized equations in the form x' = A*x

where `A` is a 4x4 coefficient matrix and `x` is the 4x1 state matrix `[dX; dTHETA; dU1; dU2]`.
Result:

```
A = [       0,                 0,          1,          0;
            0,                 0,          0,          1;
     0.098*K1,  0.72 + 0.098*K2, 0.098*K3,  0.098*K4;
      0.15*K1,  16.0 +  0.15*K2,  0.15*K3,   0.15*K4 ]
```

- Show that if `K1=K2=K3=K4=0`, then `X=THETA=U1=U2=U1'=U2'=0` is an unstable solution.
  Result: The output `ROOTS1 = EIG( evaluate(A, K1=0, K2=0, K3=0, K4=0) )` is
  `ROOTS1 = [-4; 0; 0; 4]`
  Since one of the eigenvalues is positive, the solution is unstable.
- Show that if `K1=1, K2=-244, K3=5.4, K4=-59`, then `X=THETA=U1=U2=U1'=U2'=0` is a stable solution.
  Result: The output from `ROOTS2 = EIG( evaluate(A, K1=1, K2=-244, K3=5.4, K4=-59) )` is
  `ROOTS2 = [-3.8 - 1.3*i; -3.8 + 1.3*i; -0.22 - 0.2*i; -0.22 + 0.2*i]`
  Since all of the eigenvalues have negative real parts, the solution is stable for "small" disturbances.
- Using the linearized and nonlinear equations of motion, simulate the motion of the system for `K1=1`, `K2=-244`, `K3=5.4`, `K4=-59`. Use the initial values `X=dX=1 m`, `THETA=dTHETA=20°`, `U1=dU1=0`, `U2=dU2=0`, and simulate from `T=0` to `T=20` sec. Plot `X,dX,THETA,dTHETA vs. time` and `F vs. time`.
  Result: See the plots.




Time Histories of THETA and dTHETA
THETA (DEG)
DTHETA (DEG)
THETA, dTHETA (degrees)
Time (seconds)
20
15
10
5
0
-5
0
5
10
15
20


Time History of F
F (newtons)
Time (seconds)
10
0
-10
-20
-30
-40
-50
-60
-70
-80
-90
0
5
10
15
20

```
%     File:  tutor8.al    [Autolev: http://www.autolev.com]
%  Problem:  Dynamics of a Cart Carrying an Inverted Pendulum
%----------------------------------------------------------------------
%       Default Settings
Digits       2               % Significant digits
%----------------------------------------------------------------------
%       Newtonian, bodies, frames, particles, points
Newtonian    N               % Newtonian reference frame
Bodies       A, B            % Cart, inverted pendulum
Points       O               % Point fixed in N
%----------------------------------------------------------------------
%       Variables, constants, and specified
Variables    U{2}'           % Generalized speeds; derivatives
Variables    X', THETA'      % Generalized coordinates; derivatives
Constants    L               % Length of rod
Constants    G               % Gravitational acceleration
Constants    K{4}            % Control gains
Specified    F = K1*X + K2*THETA + K3*U1 + K4*U2
%----------------------------------------------------------------------
%       Mass and inertia
Mass         A=MA, B=MB
Inertia      B, IB=MB*L^2/12, 0, IB
%----------------------------------------------------------------------
%       Geometry relating unit vectors
Simprot(N, B, 3, THETA)
%----------------------------------------------------------------------
%       Position vectors
P_O_Ao> = X*N1>
P_Ao_Bo> = 0.5*L*B2>
%----------------------------------------------------------------------
%       Kinematical differential equations
X' = U1
THETA' = U2
%----------------------------------------------------------------------
%       Angular velocities
W_A_N> = 0>
W_B_N> = THETA'*B3>
%----------------------------------------------------------------------
%       Velocities
V_Ao_N> = DT(P_O_Ao>, N)
V2pts(N, B, Ao, Bo)
%----------------------------------------------------------------------
%       Forces
Gravity( -G*N2> )
Force_Ao> += F*N1>
%----------------------------------------------------------------------
%       Equations of motion
Zero = Fr() + FrStar()
Kane()
Zero := Evaluate( Zero, L=1, MA=10, MB=1, G=9.81 )
%**********************************************************************
%       CONTROL SYSTEM / STABILITY ANALYSIS
%**********************************************************************
%       Linearization: Perturbation variables
Variables    dU{2}'          % Perturbations of U1,U2,U1',U2'
```

```
Variables    dX', dTHETA'    % Perturbations of X,X', THETA,THETA'
%-----------------------------------------------------------------------
%       Check whether X=THETA=U1=U2=U1'=U2'=0 is a solution
CHECK = Evaluate(Zero, X=0, THETA=0, U1=0, U2=0, U1'=0, U2'=0)
%-----------------------------------------------------------------------
%       Linearize kinematical equations about nominal solution
dX'    = Taylor(X',     1, X=0:dX, THETA=0:dTHETA, U1=0:dU1, U2=0:dU2)
dTHETA'= Taylor(THETA', 1, X=0:dX, THETA=0:dTHETA, U1=0:dU1, U2=0:dU2)
%-----------------------------------------------------------------------
%       Linearize equations of motion about nominal solution
PERTURB = Taylor(Zero, 1, X=0:dX, THETA=0:dTHETA, U1=0:dU1, U2=0:dU2, &
                          U1'=0:dU1', U2'=0:dU2')
Solve( PERTURB, dU{1:2}' )
%-----------------------------------------------------------------------
%       Form matrix of peturbations and its time-derivative
Xm = [ dX ,  dTHETA ,  dU1 ,  dU2 ]    % Matrix of perturbations
Xp = [ dX';  dTHETA';  dU1';  dU2']    % Time derivative of Xm
%-----------------------------------------------------------------------
%       Form matrix A such that Xm' = A * Xm
A = D( Xp, Xm )
%-----------------------------------------------------------------------
%       Stability when F=0
ROOTS1 = Eig( evaluate(A, K1=0, K2=0, K3=0, K4=0) )
%-----------------------------------------------------------------------
%       Stability when F = X - 244*THETA + 5.4*U1 - 59*U2
ROOTS2 = Eig( evaluate(A, K1=1, K2=-244, K3=5.4, K4=-59) )
%-----------------------------------------------------------------------
%       Units system for CODE input/output conversions
UnitSystem  kg,meter,sec
%-----------------------------------------------------------------------
%       Input constants, variables, etc. for CODE
Input  K1=1,  K2=-244,   K3=5.4,  K4=-59
Input   X=1 m,  THETA=20 deg,  U1=0 m/sec,  U2=0 rad/sec
Input  dX=1 m, dTHETA=20 deg, dU1=0 m/sec, dU2=0 rad/sec
Input  tInitial=0,     tFinal=20             % Begin/end times
Input  integStp=0.1,   printInt=1            % Integration/print step
Input  absErr=1.0E-08, relErr=1.0E-08        % Error tolerances
%-----------------------------------------------------------------------
%       Quantities to be output from CODE
Output T, X meters,  dX meters,  THETA deg,  dTHETA deg, F newtons
%-----------------------------------------------------------------------
%       C code generation for numerical solution
Digits  5                       % Significant digits in output files
CODE Dynamics() tutor8.c
%-----------------------------------------------------------------------
%       Record Autolev responses
Save tutor8.all
```

5.9 Spin Stabilization of a Gyrostat



The above sketch shows a gyrostat G consisting of a carrier A and a thin uniform cylindrical rotor B moving in a Newtonian frame N. Dextral sets of mutually perpendicular unit vectors Ai> (i=1,2,3) are fixed in A and are parallel to central principal axes of G. The rotor B has a central moment of inertia of J about its symmetric axes, which is parallel to A3>. The central principal moments of inertia of G for A1>, A2>, A3>, are denoted I1, I2, I3, respectively. The angle PHI measures the angle between A3> and the inertial angular momentum of G. The generalized speeds Ui (i=1,2,3) are the A1>, A2>, A3> measure numbers of the angular velocity of A in N, and the constant OMEGA is the A3> measure number of the angular velocity of B in A.

- Form equations of motion which govern angular motions of the system.
  Result: The output from the KANE command is

```
ZERO[1] = -U2*(J*OMEGA-(I2-I3)*U3) - I1*U1'
ZERO[2] = U1*(J*OMEGA-(I1-I3)*U3) - I2*U2'
ZERO[3] = (I1-I2)*U1*U2 - I3*U3'
```

- After introducing the constant nU3, show that U1=U2=U1'=U2'=U3'=0, U3=nU3 is a solution of the equations of motion.
  Result: The output from CHECK = EVALUATE(ZERO, U1=0, U1'=0, U2=0, U2'=0, U3=nU3, U3'=0) is CHECK = [0; 0; 0]

- After introducing the variables dUi and dUi' as perturbations of Ui and Ui' (i=1,2,3), respectively, linearize the equations of motion in the perturbations about the nominal solution U1=U2=U1'=U2'=U3'=0, U3=nU3. Put the linear equations in the form x' = A*x where A is a 3x3

coefficient matrix and x is the 3x1 state matrix [dU1; dU2; dU3].

Result:

```
A = [                        0, -(J*OMEGA-(I2-I3)*NU3)/I1,                0;
     (J*OMEGA-(I1-I3)*NU3)/I2,                           0,                0;
                              0,                           0,                0 ]
```

- With J=0.07634, I1=1.25, I2=4.25, I3=5, nU3=1, determine values of OMEGA which result in LAMBDA, the eigenvalues of A to be positive.

  Result: The Autolev output response to the last line of the file tutor9.al is

  ```
  DET = 0.001096997*(9.824469+OMEGA)*(49.12235+OMEGA) + LAMBDA^2
  ```

  This shows that LAMBDA^2 is positive when -49.12 < OMEGA < -9.82 . When the real part of LAMBDA is positive, the system is said to be "unstable". When the real part of LAMBDA is 0, the system is said to be "neutrally stable". Hence, the gyro is neutrally stable for OMEGA > -9.82 and for OMEGA < -49.12.

- Using the nonlinear equations of motion, run four numerical motion simulations for $0 \leq t \leq 20$ sec. In all four simulations, set J=0.07634, I1=1.25, I2=4.25, I3=5, U1=0, and U3=1.

  In the first, set OMEGA=-7 and U2=.02. In the second, set OMEGA=-7 and U2=.01.

  In the third, set OMEGA=-20 and U2=.02. In the fourth, set OMEGA=-20 and U2=.01.

  Plot PHI versus time. For each simulation, check that H, the magnitude of the inertial angular momentum of the gyrostat, is time-invariant.

  Result: See the plots. By examining the simulation output file tutor9.1, it is clear that H is time-invariant.


Time-histories of PHI when OMEGA=-7 rad/sec
PHI (degrees)
Time (seconds)
U2=.01 rad/sec
U2=.02 rad/sec
0
0.2
0.4
0.6
0.8
1
1.2
0
5
10
15
20


Time-histories of PHI when OMEGA=-20 rad/sec
PHI (degrees)
Time (seconds)
U2=.01 rad/sec
U2=.02 rad/sec
0
20
40
60
80
100
120
140
0
5
10
15
20

```
%     File:  tutor9.al    [Autolev: http://www.autolev.com]
%  Problem:  Spin stability of gyrostat
%--------------------------------------------------------------------
%       Newtonian, bodies, frames, particles, points
Newtonian    N               % Newtonian reference frame
Bodies       A               % Carrier
Frames       B               % Rotor
%--------------------------------------------------------------------
%       Variables, constants, and specified
Variables    U{3}'           % Generalized speeds; derivatives
Constants    OMEGA           % Angular speed of B in A
Constants    J               % Moment of inertia of B about spin axis
%--------------------------------------------------------------------
%       Mass and inertia
Mass         A=M             % Attribute mass of G to A
Inertia      A, I1, I2, I3   % Attribute inertia of G to A
%--------------------------------------------------------------------
%       Angular velocities
W_A_N> = U1*A1> + U2*A2> + U3*A3>
W_B_A> = OMEGA*A3>
%--------------------------------------------------------------------
%       Angular accelerations
ALF_A_N> = DT(W_A_N>, N)
ALF_B_A> = DT(W_B_A>, A)
%--------------------------------------------------------------------
%       Acceleration of mass center of G is 0>
A_Ao_N> = 0>
%--------------------------------------------------------------------
%       Equations of motion
Zero = Fr() + FrStar() + Gyrostat(FrStar,CYLINDER,A,B,J)
Kane()
%--------------------------------------------------------------------
%       Angular momentum of gyrostat
H> = Momentum(ANGULAR, Ao) + Gyrostat(ANGMOM,CYLINDER,A,B,J)
H = Mag(H>)
%--------------------------------------------------------------------
%       Angle between angular momentum vector and A3>
PHI = acos( Dot( UNITVEC(H>), A3>) )
%--------------------------------------------------------------------
%       Units system for CODE input/output conversions
UnitSystem  kg,meter,sec
%--------------------------------------------------------------------
%       Input constants, variables, etc. for CODE
Input   tInitial=0,     tFinal=20            % Begin/end times
Input   integStp=0.1,   printInt=1           % Integration/print step
Input   absErr=1.0E-07, relErr=1.0E-07       % Error tolerances
Input   J=0.07634, I1=1.25, I2=4.25, I3=5
```

```
Input   OMEGA=-20.0, U1=0, U2=0.02, U3=1
%-----------------------------------------------------------------
%       Quantities to be output from CODE
Output  T, PHI degs, H
%-----------------------------------------------------------------
%       Fortran code generation for numerical solution
CODE Dynamics() tutor9.for
%*****************************************************************
%       STABILITY ANALYSIS
%*****************************************************************
%       Linearization: Perturbation vars + nominal solution parameters
Variables    dU{3}'          % Perturbations of U1, U2, U3
Constants    nU3             % Nominal solution for U3
%-----------------------------------------------------------------
%       Check nominal solution
CHECK = Evaluate(Zero, U1=0, U1'=0, U2=0, U2'=0, U3=nU3, U3'=0)
%-----------------------------------------------------------------
%       Linearize equations of motion about nominal solution
PETURB  = Taylor(Zero, 1, U1=0:dU1, U1'=0:dU1', U2=0:dU2, U2'=0:dU2', &
                          U3=nU3:dU3, U3'=0:dU3')
Solve( PETURB,  dU{1:3}' )
%-----------------------------------------------------------------
%       Form, x, x', and the A matrix in the equation x'=A*x
Xm = [ dU1,  dU2,  dU3 ]
Xp = [ dU1'; dU2'; dU3']
Am = D( Xp, Xm )
%-----------------------------------------------------------------
%       To find eigenvalues of Am symbolically, find the roots of the
%       equation found by setting the determinant of LAMBDA*I-A = 0
Variables LAMBDA
DET = Det( LAMBDA*DIAGMAT(3,1) - Am )
%-----------------------------------------------------------------
%       Inspection of DET shows that LAMBDA=0 is a root
DET /= LAMBDA
%-----------------------------------------------------------------
%       With J=0.07634, I1=1.25, I2=4.25, I3=5, nU3=1,
%       determine the values of OMEGA which result in LAMBDA > 0
DET := Evaluate(DET, J=0.07634, I1=1.25, I2=4.25, I3=5, nU3=1)
%-----------------------------------------------------------------
%       Record Autolev responses
Save tutor9.all
```

6 Other Information

6.1 Functions and Commands

`Autolev` is equipped with more than 100 commands. An alphabetical list of commands appears on the screen when one types `WHAT` at any `Autolev` line prompt. A Quick-Reference and a Summary of Commands are included at the end of the tutorial. In addition, a detailed description of a command appears on the screen when one types `HELP commandname` at any `Autolev` line prompt.

Most commands may be nested. For example, the `DOT` command may appear as an argument of the `ACOS` command, e.g., `ANGLE = ACOS( DOT(A1>,B1>) )`.

6.2 Stand-Alone Commands

In many commands, e.g., `RENEE = COS(X)`, one uses an equals sign to make an assignment. In some commands, called *stand-alone commands*, one makes assignments without using equals signs. For example, the equations

```
3*X + 4*Y = 37
4*X - 2*Y = -2
```

can be solved by executing the following `Autolev` input file

```
   (1) Variables X, Y
   (2) Zero[1] = -37 + 3*X + 4*Y
-> (3) Zero[1] = -37 + 3*X + 4*Y
   (4) Zero[2] = 2 + 4*X - 2*Y
-> (5) Zero[2] = 2 + 4*X - 2*Y
   (6) SOLVE( Zero, X,Y )
-> (7) X = 3
-> (8) Y = 7
```

In line 6, the command `SOLVE(Zero,X,Y)` causes values to be assigned to `X` and `Y`, but the command does not involve the typing of any equals sign. Some other stand-alone commands are `V2pts`, `V1pt`, `A2pts`, and `A1pt`.

6.3 Dual Functions

The commands ARRANGE, EXPAND, EXPLICIT, EXPRESS, FACTOR, and ZEE are called *dual functions* because they can be used in two ways, namely, on the right-hand side of an equals sign, e.g., NEWY = EXPLICIT(Y), or as stand-alone commands. When a dual-function appears on an Autolev input line in the form

```
DualFunctionName( X, list_of_arguments )
```

where DualFunctionName is one of the dual functions, X is the *name* of an expression, not the expression itself, and list_of_arguments stands for arguments of the dual function, then Autolev interprets this line as

```
X := DualFunctionName( X, list_of_arguments )
```

Dual functions differ from other commands in that they do not change the functional character of an expression, but alter its appearance.

6.4 Creating Your Own Commands: .A and .R Files

To add to Autolev's more than 100 commands, you can create new commands by creating an ASCII file that resides in the Autolev toolbox directory, the current working directory, or a directory that is specified in the AUTOLEVPATH. For example, suppose you wish to create a command called SUM, which adds two expressions and returns their sum. The syntax of the SUM command could be SUM(x,y), where x and y are expressions. To create the command SUM, use a text editor to compose a file named sum.r with the following contents:

```
%SUM.R
%
%Function:  Returns the sum of two expressions.
%
%  Syntax:  SUM(x,y)
%
#1# + #2#
```

The filename has a .r extension to inform Autolev that SUM returns a result (.r for result). The #1# and #2# denote the two arguments of SUM. The comments at the top of the file contain text that appears on the screen when HELP SUM is typed at the Autolev line prompt. Typing PLUS = SUM(3,5) results in PLUS = 8 .

Next, suppose you want to create a command which makes assignments. For example, to create a command called POWER, which squares and cubes an expression and assigns the results to new variables called *name*SQ and *name*CUBE, compose a file named power.a with the following contents:

```
%POWER.A
%
%Function:  Put explanatory information about POWER here.
%
%Syntax:    POWER(expression,name)
%
#2#SQ = (#1#)^2
#2#CUBE = (#1#)^3
```

The filename has a .a extension to inform `Autolev` that `POWER` makes assignments (.a for assignment). Note the use of parentheses around the `#1#` argument. Liberal use of parentheses is helpful in making bug-free .a and .r files. To produce efficient results when `AUTOZ` is `ON`, use the `AUTOZ()` command which introduces intermediate symbols. Use of the `POWER` command is demonstrated below.

```
   (1) POWER(3,A)
 ->(2) ASQ = 9
 ->(3) ACUBE = 27
   (4) Constants A,B
   (5) POWER(A+B,ED)
 ->(6) EDSQ = (A+B)^2
 ->(7) EDCUBE = (A+B)^3
```

There are three special symbols that are helpful in creating .a and .r files. The symbol `#NUM_ARGS#` evaluates to the number of arguments passed to the .a or .r command; `#Newtonian#` evaluates to the name declared in the Newtonian declaration; and `#DEGREES#` evaluates to ON or OFF, depending on the current setting of DEGREES. In addition, one may use the logical `if` and `else` statements, the `ECHO` command, and the special symbols `\k`, `\a`, `\p`, `\n` that are used with `ECHO` (type `HELP ECHO` for more information). For example,

```
%SUM.R
%
%Function:  Returns the sum of two expressions.
%
%  Syntax:  SUM(x,y)
%
if( #NUM_ARGS# != 2 )
   { ECHO(\k\a"Error: wrong number of arguments to the SUM command"\p\n); }
else
   { #1# + #2# }
```

6.5 Default Settings

The manner in which `Autolev` responds to online commands or when executing a batch file depends on certain "settings". For example, the factoring of expressions depends on whether `FACTORING` is `ON` or `OFF`, as seen by comparing the two `Autolev` sessions below:

```
   (1) FACTORING OFF              (1) FACTORING ON
   (2) Constants A,B,C            (2) Constants A,B,C
   (3) X = A*B + A*C + A^2        (3) X = A*B + A*C + A^2
-> (4) X = A*B + A*C + A^2     -> (4) X = A*(A+B+C)
   (5) Y = A/C + B/C              (5) Y = A/C + B/C
-> (6) Y = A/C + B/C           -> (6) Y = (A+B)/C
```

AUTORHS is another example of a settting. AUTORHS may be set to ALL, ON, or OFF. With AUTORHS set to ALL, the right-hand side of all quantities are automatically substituted for the quantity. With AUTORHS ON, substitutions are made when the right-hand side of quantities are "simple". With AUTORHS OFF, no substitutions are made. One way to see the effect of AUTORHS is to compare the files below.

```
   (1) AUTORHS  OFF          (1) AUTORHS  ON           (1) AUTORHS  ALL
   (2) Constants B           (2) Constants B           (2) Constants B
   (3) A = 7                 (3) A = 7                 (3) A = 7
-> (4) A = 7              -> (4) A = 7              -> (4) A = 7
   (5) C = A + B             (5) C = A + B             (5) C = A + B
-> (6) C = A + B          -> (6) C = 7 + B          -> (6) C = 7 + B
   (7) D = C*SIN(C)          (7) D = C*SIN(C)          (7) D = C*SIN(C)
-> (8) D = C*SIN(C)       -> (8) D = C*SIN(C)       -> (8) D = (7+B)*SIN(7+B)
```

Whenever Autolev is invoked, several settings are assigned default values in the file DEFAULTS.AL. The user can employ a text editor to change each of the values in this file. For example, ON can be replaced with OFF as the value assigned to FACTORING. The value assigned to most of the settings in the file DEFAULTS.AL can be changed online. For example, if the value assigned in DEFAULTS.AL to AUTORHS is ON, then entering the line AUTORHS OFF changes the setting from ON to OFF in the current Autolev session. Such changes can be made repeatedly. However, changes online do not alter the file DEFAULTS.AL.

During an Autolev session, one can review the values assigned to all settings by typing DEFAULTS at any line prompt. For help with a particular setting, type HELP SettingName, e.g., HELP FACTORING.

Note: To accommodate multi-user computers, Autolev checks the current directory for the file DEFAULTS.AL before checking the directories specified in the environment variable AUTOLEVPATH or the TOOLBOX directory or the directories specified in the environment variable PATH.

6.6 Special Symbols (see also Reserved Names in Section 2.4)

%	Comment delimiter. Input following this character is ignored
%%	Inserts a comment in MATLAB, C, or FORTRAN code
&	Line continuation character (Autolev input files only)
' (prime)	Implies total differentiation with respect to T
>	The last symbol in the name of a vector
>>	The last two symbols in the name of a dyadic
>>>	The last three symbols in the name of a triadic or higher order polyadic
[ ]	Used to denote a matrix or designate an element of a matrix
{ }	Encloses indices in declarations
,	Separates arguments of a function and elements of a row in a matrix
;	Separates rows of a matrix
:	Designates a range, e.g., 1:3
()	Encloses mathematical expressions and function arguments
#	Delimits arguments in .A and .R files
"	Delimits string literals
.	Decimal point
+ - * / ^	Mathematical operators: addition, subtraction, multiplication, division, and exponentiation
= := += -=	Assignment operators: normal, overwrite, addition, and subtraction
*= /= ^ =	Assignment operators: multiplication, division, and exponentiation

6.7 Online Editing

With a Macintosh or workstation version of Autolev, online editing is most easily accomplished with the copy and paste commands (Option-C and Option-V on a Macintosh). When a Windows version of Autolev is used, the following keys are helpful in interactive mode:

HOME	Moves cursor to the beginning of input line
END	Moves cursor to the end of input line
DEL	Deletes current character in line
BACKSPACE	Deletes previous character in line
RIGHT ARROW →	Moves cursor one space to the right
LEFT ARROW ←	Moves cursor one space to the left
UP ARROW ↑	Recovers previous line
INSERT	Toggles between insert and overstrike editing modes
Control-C	Abruptly terminates program execution

7 Summary of Commands

Interfacing with Autolev and the operating system

`al`	Invokes `Autolev` from the operating system prompt
`al filename`	Invokes `Autolev` from the operating system prompt and executes the commands in `filename`
`CLEAR`	Deletes all lines from the workspace and restarts `Autolev` at line (1)
`CLEAR k`	Deletes lines k and all lines following line k from the workspace and reruns `Autolev` from line(1) to line (k-1)
`CONTROL-C`	Terminates program execution
`ECHO`	Enables information to be transmitted to the user during execution of user-created .a and/or .r files
`EXIT or QUIT`	Ends `Autolev` session
`HELP`	Displays general information about `Autolev` and OnLine Dynamics, Inc.
`HELP commandname`	Displays information on commands, declarations, and defaults
`HELP syntax_forms`	Displays information on syntactical forms
`HELP update`	Displays information on new commands, declarations, and defaults
`LIST`	Displays commands and responses beginning with line (1)
`LIST 10:20`	Displays lines 10 through 20 on the screen
`MEMORY`	Displays memory allocation information
`PAUSE`	Temporarily suspends execution
`RUN`	Continues execution of an `Autolev` input file after an error occurs
`RUN filename`	Executes the `Autolev` commands listed in `filename`
`SAVE filename.al`	Creates `filename.al` containing all input lines
`SAVE filename.all`	Creates `filename.all` containing all input lines and response lines
`WHAT`	Displays a list of commands, declarations, and defaults
`!`	Suspends program execution and invokes operating system
`!abcd`	Issues operating system command `abcd` from within `Autolev`

Default settings

`AutoEpsilon 1.0E-7`	A number differing from an integer by `1.0E-7` or less is replaced with this integer. `AutoEpsilon` is useful when tiny deviations of numbers from their neighboring integer values adversely affects `Autolev`'s ability to simplify expressions
`AutoExpress on/off`	Causes linear/angular velocities and linear/angular accelerations to be expressed in terms of certain basis vectors
`AUTOLEVPATH`	Controls the search on disk for files used by `Autolev`. `AUTOLEVPATH` is an environment variable which must be set in the file `AUTOEXEC.BAT` (Windows 95/98/ME) or in the file `.cshrc` or `.login` (Unix, Linux)
`AutoRHS on/off/all`	Controls substitution for the right-hand sides of equations. See also `HELP EXPLICIT` and `HELP RHS`. Note: Normally, `AutoRHS` should not be set to `ALL` because this produces computationally inefficient formulas
`AutoZ on/off`	Controls automatic introduction of intermediate variables `Z1, Z2, ...` For large problems or to produce computationally efficient formulas, set `AutoZ ON`
`COMPLEX on/off/auto`	Controls some simplification of scalar quantities. Setting `COMPLEX ON` tells `Autolev` to treat scalar quantities as complex, whereas setting `COMPLEX OFF` allows `Autolev` to assume that scalar quantities have no imaginary parts. Setting `COMPLEX AUTO` tells `Autolev` that `COMPLEX` is `OFF` until `Autolev` encounters an imaginary number
`DEFAULTS`	Lists current settings of all defaults
`DEGREES on/off`	Determines whether the arguments of trigonometric functions (e.g., `SIN`, `COS`, `TAN`) and the value returned by inverse trigonometric (e.g., `ASIN`, `ACOS`, `ATAN`, and `ATAN2`) are regarded as representing degrees or radians. The setting of `DEGREES` does *not* affect degrees and radians conversion in `Matlab`, C, or Fortran code. See also `HELP UnitSystem`
`DIGITS n`	Controls the number of digits displayed for online calculations and in output files from `Autolev` generated `Matlab`, C, or Fortran code to be `n` ($1 \leq n \leq 17$)
`FACTORING on/off`	Controls the automatic factoring of expressions
`Imaginary i`	Declares `i` as the imaginary number, $\sqrt{-1}$
`OVERWRITE on/off`	Controls whether users are queried when quantities are overwritten. Note: Quantities may also be overwritten, without query, with :=, the overwrite assignment operator
`PAUSE wait,0,1,..`	Controls delay time when a warning is issued
`SCRATCH_DIRECTORY`	Controls the location on disk of scratch files. `SCRATCH_DIRECTORY` can be set only in the file `defaults.al` and may be set to an empty string, the name of a directory, or, for fastest execution, a virtual RAM-drive
`SIMPLIFY`	Controls simplification of certain mathematical expressions involving inverse trigonometric functions
`SPACING 0,1,...`	Controls inter-line spacing
`STEPPING`	Enables execution of an input file one line at a time

Physical declarations

`Bodies A`	Declares the (massive) body `A`, the point `Ao` (mass center of A), and orthonormal vectors `A1>, A2>, A3>` fixed in `A`
`Frames B`	Declares the reference frame `B`, the point `Bo`, and orthonormal vectors `B1>, B2>, B3>` fixed in `B`
`Newtonian N`	Declares `N` to be a Newtonian (inertial) reference frame
`Particles C,D`	Declares the (massive) particles `C` and `D`
`Points E,F`	Declares the (massless) points `D` and `E`

Mathematical declarations

`Constants a,b+,c-`	Declares the constant `a`, the non-negative constant `b`, and the non-positive constant `c`
`Imaginary j`	Declares `j` as the imaginary number, $\sqrt{-1}$
`Specified f`	Declares `f` as a specified function of constants, time, and/or other variables
`Variables x'`	Declares the variable `x` and its first time-derivative `x'`

Mass and inertia declarations

`Inertia B,I1,I2,I3,I12,I23,I31`	Declares the central inertia scalars of body B
`Mass B=mB`	Declares `mB` as a non-negative constant and assigns it to the mass of the particle or body `B`

Mathematical operators and library functions

`+ - * / ^`	Mathematical operators for addition, subtraction, multiplication, division, and exponentiation
`+= -= *= /= ^=`	Cumulative mathematical operators for addition, subtraction, multiplication, division, and exponentiation
`abs(x)`	Absolute value
`cos(x) sin(x) tan(x)`	Trigonometric functions
`cosh(x) sinh(x) tanh(x)`	Hyperbolic functions
`acos(x) asin(x) atan(x)`	Inverse trigonometric functions
`atan2(y,x)`	Two argument inverse tangent function
`log(x) log10(x)`	Natural logarithm and base 10 logarithm
`exp(x)`	Exponential function
`sqr(x) sqrt(x)`	Square or square-root function
`factorial(x)`	Factorial of non-negative integer
`ceil(x) floor(x)`	Rounds floating point number up or down to nearest integer
`int(x) round(x)`	Truncates or rounds floating point number to integer
`max(x,y) min(x,y)`	Returns the maximum or minimum of two numbers
`sign(x)`	Returns +1 if $x > 0$, 0 if x=0, or -1 if $x < 0$

Mathematical commands

`AutoTaylor(0:2,w=a,x=0)`	Governs automatic Taylor series expansion of expressions about `w=a,x=0`. Useful for linearization
`D(y,x)`	Returns the partial derivative of `y` with respect to `x`
`Dt(y)`	Returns the total derivative of `y` with respect to `t` (time)
`Evaluate(y,a=x,b=2)`	Returns the expression for `y` evaluated with `a=x` and `b=2`
`Polynomial(y,x,n)`	Returns a matrix whose elements are the coefficients of $x^n, x^{n-1}, ..., x^1, x^0$ in `y`
`Roots(A)`	Returns the roots of the polynomial whose coefficients are stored in the matrix `A`
`Roots(y,x,n)`	Returns an $n \times 1$ matrix of roots of `y`, where `y` is a polynomial in `x`
`Solve(A,x1,x2)`	Solves for the variables `x1` and `x2` that appear linearly in the equations represented by the matrix `A`
`Solve(option,A,X)`	Solves the linear equations in matrix `A` for the variables in matrix `X`. `Option` is `Gauss`, `Minors` or `Implicit`
`Spline(...)`	Returns a step, bell, transition, pulse, or spline function
`Taylor(y,0:2,w=a,x=0)`	Returns the 0th, 1st, and 2nd degree terms of the Taylor series expansion of `y` about `w=a,x=0`.
`Units(n*m,lbf*ft)`	Returns the conversion factor between newton*meters and pound-force*feet

Vector and dyadic commands

`+ - *`	Vector and dyadic operators
`0>  1>>`	Zero vector. Unit dyadic
`Cross(a>,b>)`	Returns the cross-product of the vector `a>` and the vector `b>`
`Dot(a>,b>)`	Returns the dot-product of the vector `a>` and the vector `b>`
`Dot(a>,b>>)`	Returns the dot-product of the vector `a>` and the dyadic `b>`
`Dyadic`	Constructs a dyadic from a frame (or body) and scalars or a matrix of scalars
`Express(v>,B)`	Expresses `v>` in terms of the unit vectors `B1>, B2>, B3>` fixed in body (or frame) `B`
`Mag(v>)`	Returns the magnitude of `v>`
`Magsq(v>)`	Returns the square of the magnitude of `v>`, frequently denoted by $\mathbf{v}^2$. Note: `MAGSQ(v>)` $\triangleq$ `DOT(v>,v>)`
`Trace(d>>)`	Returns the trace of the dyadic `d>>`
`Unitvec(v>)`	Returns a unit vector having the same direction as `v>`
`Vector`	Constructs a vector from a frame (or body) and three scalars or a matrix of three scalars

Matrix commands

`+ - *`	Matrix operators
`Cols(A)`	Returns the number of columns in matrix `A`
`Cols(A,1,3,4)`	Returns a matrix consisting of the 1^{st}, 3^{rd}, and 4^{th} columns of the matrix `A`
`Det(A)`	Returns the determinant of the square matrix `A`
`DiagMat(n,m,x)`	Returns the $n \times m$ matrix with `x` along the "diagonal" and all other elements equal to zero
`Eig(A)`	Returns a column matrix whose elements are the eigenvalues of `A`
`Eig(A,LAMBDA,V)`	Assigns the eigenvalues of `A` to the column matrix `LAMBDA`, and the eigenvectors of A to the *rows* of `V`
`Element(A,2,3)`	Returns A[2,3], the element in the 2^{nd} row and 3^{rd} column of matrix `A`
`Inv(A)`	Returns the inverse of the square matrix `A`
`Matrix`	Constructs a matrix from a reference frame and a vector, dyadic, or triadic
`Polynomial(A,x)`	Returns a polynomial in x whose coefficients are elements of the row or column matrix `A`
`Rows(A)`	Returns the number of rows in matrix `A`
`Rows(A,1:3,5:7)`	Returns a matrix consisting of the 1^{st}-3^{rd} and 5^{th}-7^{th} rows of the matrix `A`
`Trace(A)`	Returns the trace (sum of the elements on the "diagonal") of the matrix `A`
`Transpose(A)`	Returns the transpose of matrix `A`

Mass distribution commands

`CM(P)`	Returns the position vector from point `P` to the mass center of a system of particles and/or bodies
`Inertia(P,A,B)`	Returns the sum of the inertia dyadics of `A` and `B` for point `P`
`Mass(A,B,C)`	Returns the sum of the masses of `A`, `B`, and `C`

Kinematics commands

`A1pt(A,B,Bq,Q)`	Forms `A_Q_A>` by implementing the acceleration formula for one point moving on a rigid body
`A2pts(A,B,P,Q)`	Forms `A_Q_A>` by implementing the acceleration formula for two points fixed on a rigid body
`Angvel(A,B,...)`	Forms `W_B_A>`, the angular velocity of `B` in `A` associated with Euler angles, Rodrigues parameters, or Euler parameters
`Dircos(A,B,...)`	Forms the `A_B` direction cosine matrix associated with Euler angles, Rodrigues parameters, or Euler parameters
`Kindiffs(A,B,...)`	Forms kinematical differential equations for Euler angles, Rodrigues parameters, or Euler parameters
`Partials(V_P_N>)`	Returns a matrix of partial velocities of `P` in `N`
`Simprot(A,B,1,x)`	Forms the `A_B` direction cosine matrix associated with a simple rotation of amount `x` about `A1> or B1>`
`Remainder(V_P_N>)`	Returns the velocity remainder of `P` in `N`
`V1pt(A,B,Bq,Q)`	Forms `V_Q_A>` by implementing the velocity formula for one point moving on a rigid body
`V2pts(A,B,P,Q)`	Forms `V_Q_A>` by implementing the velocity formula for two points fixed on a rigid body

Kinetics commands

`Gravity(G*n1>)`	Adds a local gravitational force to each particle and body
`Force(P/Q,v>)`	Adds -v> to the force on `P`, adds +v> to the force on `Q`
`Fr()`	Returns a matrix of generalized active forces
`Torque(A/B,v>)`	Adds -v> to the torque on `A`, adds +v> to the torque on `B`

Dynamics commands

`Constrain(...)`	Solves for dependent and auxiliary generalized speeds and their time derivatives
`FrStar()`	Returns a matrix of generalized inertia forces
`Gyrostat(option)`	Facilitates formulation of expressions for `FRSTAR`, `KE`, and `ANGMOM` for gyrostats
`Kane()`	Brings Kane's dynamical equations into a form suitable for numerical integration; determines force/torque measure numbers associated with auxiliary generalized speeds
`KE()`	Forms kinetic energy
`Momentum(option)`	Forms `LINEAR`, `ANGULAR`, and `GENERALIZED` momentum
`NiCheck()`	Forms a function that should remain constant throughout numerical integration of equations of motion
`TStar(B,N)`	Forms inertia torque of body `B` in frame `N`

Code commands

`Animate`	Informs Autolev that certain frames and/or bodies are to be animated by means of the program Animake
`CODE Option()`	Generates a `Matlab`, C, or Fortran program. `Option` is `Dynamics`, `ODE`, `Nonlinear`, or `Algebraic`
`Digits`	Governs the number of digits displayed in output files of `Matlab`, C, or Fortran programs
`Input X=2,Y=3`	Assigns the value 2 to `X` and 3 to `Y` for use in a `Matlab`, C, or Fortran input file
`Input X=2 deg, Y=3 m`	Assigns the value 2 degrees to `X` and 3 meters to `Y` for use in a `Matlab`, C, or Fortran input file
`Input(X)`	Returns the input value assigned to `X`
`Input(X,UnitSystem)`	Returns the input value assigned to `X` multiplied by `conversionFactor`, where `conversionFactor` is the conversion factor from the units assigned to `X` to the units named in the UnitSystem declaration
`Output T,X+Y`	Specifies `T` and `X+Y` as output quantities for `Matlab`, C, or Fortran programs
`Output T sec, X+Y m`	Specifies `T` in seconds and `X+Y` in meters as output quantities for `Matlab`, C, or Fortran programs
`UnitSystem`	Declares a unit system for automatic conversion of units declared in Input or Output commands

Simplifications commands

`Arrange(y,n,x) *`	Arranges in groups those terms in `y` which are of degree `n` in `x` (`n` = 0, 1, or 2)
`AutoZ(y,x)`	If AutoZ is ON, returns `ZEE(y,x)`, else returns `y`
`Coef(y,x)`	Returns the coefficient of `x` in the expression for `y` (`y` must be linear in `x`)
`Epsilon(y,0.1)`	Returns an expression found by setting numbers in `y` which are within `0.1` of an integer equal to that integer
`Exclude(y,x)`	Returns the terms in `y` that do not contain `x` (`y` does *not* have to be linear in `x`)
`Expand(y,n:m) *`	Removes parentheses, e.g., `(a+b)*c` becomes `a*c + b*c`
`Explicit(y,x)`	Makes `y` an explicit function of `x`; e.g., if `a=x+t` and `y=a+t`, `explicit(y,t)` produces `x+2*t`
`Factor(y,x) *`	Returns `y` factored on `x`
`Include(y,x)`	Returns the terms in `y` that contain `x` (`y` must be linear in `x`)
`Replace(y,sin(x)=3)`	Replaces `sin(x)` with 3 in the expression for `y`
`RHS(y)`	Returns the right-hand side of `y`
`Zee(y) *`	Produces a computationally efficient expression for `y` by introducing intermediate variables `Z1, Z2, ....`
`Zee(y,x) *`	Produces a compact expression for `y` by introducing variables `Z1, Z2, ...` which do not explicitly contain `x`. (`y` must be linear in `x`)

* denotes dual function, i.e., a command that can be entered either on the right-hand side of an equation or immediately following an Autolev prompt.